

# Einführung in Perl

Version 0.04

Email: <mailto:martin@fabiani.net>

Homepage: <http://www.fabiani.net>

Copyright by: Martin Fabiani 2003

Dieses Dokument darf für den Eigenbedarf gelesen und abgespeichert werden!

Dieses Dokument darf nicht ohne meine Erlaubnis veröffentlicht werden!

Dieses Dokument darf nicht verändert werden!

Dieses Dokument darf nicht ohne meine Zustimmung für kommerzielle Zwecke wie Schulungen oder ähnliches verwendet werden!

Bei Fragen: Email an: <mailto:martin@fabiani.net>

# Vorwort

*The very fact that it's possible to write messy programs in Perl is also what makes it possible to write programs that are cleaner in Perl than they could ever be in a language that attempts to enforce cleanliness.*  
(Larry Wall, der Vater von Perl, Linux World 1998)

## Warum Perl?

1. Perl hat sehr mächtige Möglichkeiten zur Behandlung von Zeichenketten. Da sich sehr viele Problemstellungen daraufhin zurückführen lassen, sind die Anwendungsmöglichkeiten für Perl sehr groß, z.B. mächtigere Shellscripate, Logfile-Auswertungen, Webanwendungen, XML-Transformationen, Web-Spider, ...
2. Perl-Module bieten Schnittstellen zu sehr vielen verschiedenen Systemen, also kann man Perl meistens sehr gut verwenden, um diese zu verbinden oder zu automatisieren, z.B. Datenbanken, Directories, Excel ;-), Telnet, FTP, SSH, Fernsteuerung von Webanwendungen über LWP, zu anderen Programmiersprachen wie C (h2xs, Inline::C), C++, Java, ... Perl wird aus diesem Grund auch "Glue-Language" genannt, weil sie viele verschiedenen Systeme "zusammenklebt".
3. Bei sehr vielen Problemstellungen hat man eine sehr kurze Entwicklungszeit, und man kann häufig entscheiden, auf welcher Ebene man ein Problem angehen will (z.B. will ich mit einzelnen Zeichen arbeiten, oder mit Zeilen, Absätzen, kompletten Texten, ...)
4. Perl ist sehr portabel und es läuft auf sehr vielen verschiedenen Plattformen
5. Perl ist frei
6. Man findet sehr gute Hilfe im Web (z.B. <http://www.perl-community.de/>, <http://www.perlmonks.org/>)

## Haupteinsatzgebiete

- Erweiterte Shellscripate
  - Fernsteuerung von anderen Anwendungen
  - Zugriff auf Datenbanken, Directories, Internetdienste (ftp, http, mail, ...)
  - Scannen und Parsen großer Datenmengen
  - XML und dessen Kinder (smil, svg, ...)
- Anwendungen mit GUI (Tk, wxPerl, gtk, ...) oder ohne
- Web-Programmierung (Perl/CGI, mod\_perl, FastCGI, ...)

## Was bedeutet der Name Perl?

Practical Extraction and Report Language

Eine weitere Deutung ist: Pathologically Eclectic Rubbish Lister

## Wie alt ist Perl?

Perl 1.00 wurde 1987 freigegeben. 1994 wurde Perl5.0 released. Die aktuelle Version Perl5.8 wurde 2002 released.

## Wo liegen die Wurzeln von Perl?

Perl hat Anleihen aus den verschiedensten Sprachen und Tools genommen, wie C, Shell-Programmierung, LISP, Basic, sed, awk und vielen anderen. Die Syntax ist sehr C-ähnlich, aber viel freier. Perl ist übrigens keine Sprache, die auf einer Universität entstanden ist (wie Pascal und dessen Abkömmlinge), sondern wurde von einem Sprachwissenschaftler entwickelt und versucht, sich an der gesprochenen Sprache (d.h. Englisch) zu orientieren..

# Basiswissen

## Das erste Programm:

Code: *hello\_1.pl*

```
1: #!/usr/bin/perl
2:
3: print "Hello Linux-World\n";
```

In der ersten Zeile steht der "Shebang", d.h. dort wird angegeben womit dieses Script ausgeführt werden soll (analog zu Shellsripten). Nur ist es bei Perl nicht `/bin/sh` oder ähnliches, sondern `/usr/bin/perl`.

In der dritten Zeile steht ein Kommando: gebe am Bildschirm einen Text aus und danach einen Zeilenumbruch `\n` (Die Zeilennummern gehören nicht zum Code, ich füge sie jedoch hier in diesem Vortrag immer ein, damit man die Zeilen besser benennen kann, ebenso wie ich den Code farbige mache. Normales Schwarz/Weiß genügt für Perl)

Diese Zeilen könnte man mit einem (fast) beliebigen Texteditor (z.B. UltraEdit, Emacs, vi, ...) in eine Datei namens *hello\_1.pl* schreiben, (unter Linux/Unix der Datei dann Execute-Rechte geben (`chmod 755 hello_1.pl`)) und sie dann in der Shell ausführen: `./hello_1.pl` oder `perl hello_1.pl`. Daraufhin würde an der Shell die Ausgabe: *Hello Linux-World* erscheinen

Obwohl die Dateierweiterung unter Linux keine Bedeutung hat, wird für Perl-Skripte häufig die Endung `.pl` verwendet. Unter Windows hingegen werden Dateien normalerweise nach ihrer Dateierweiterung Programmen zugeordnet, deshalb wird da fast immer die Endung `.pl` verwendet.

## Skalare Variablen:

Code: *hello\_2.pl*

```
1: #!/usr/bin/perl
2:
3: $greetingMessage = "Hello Linux-World";
4:
5: print "$greetingMessage\n";
6: print $greetingMessage . "\n";
```

In dieser Variante wird die Zeichenkette einer Variablen namens `$greetingMessage` zugewiesen (Zeile 3). Variablen beginnen zunächst immer mit einem `$`-Zeichen und sie können Zeichen(-ketten) und Zahlen aufnehmen. Weil sie genau einen Wert aufnehmen können, bezeichnet man sie in Perl auch als **skalare Variablen**. Gültige Variablennamen beginnen mit einem `$` und danach einem Buchstaben oder einem Unterstrich, gefolgt von beliebig vielen anderen Buchstaben, Zahlen und Unterstrichen: `[$[A-Za-z_][A-Za-z0-9_]*`. Deutsche Umlaute sollten nicht verwendet werden, ebenso die Namen `$a` und `$b` (dazu später mehr).

- Eine skalare Variable beginnt immer mit dem Zeichen `$`
- Die Zuweisung erfolgt über das Zeichen `=` (Zeile 3)
- Skalare Variablen können auch in Zeichenketten mit `"...."` vorkommen. (Zeile 5)
- Man kann mehrere Zeichenkette mit dem `.` zusammenhängen. (Zeile 6)
- Jeder Ausdruck wird durch ein Semikolon `;` abgeschlossen (wie bei C/C++/Java/Pascal)

## Input von der Shell lesen und auswerten:

Code: *hello\_3.pl*

```
1: #! /usr/bin/perl
2:
3: print "Wie ist dein Name? ";
4: $name = <STDIN>;
5: chomp($name);
6: print "Hallo $name\n";
```

- Die Zeilen 1,3 und 6 verstehen wir schon
- In Zeile 5 wird die Funktion `chomp()` auf eine Variable namens `$name` angewendet. Diese Funktion schneidet einen Zeilenumbruch am Ende einer Zeichenkette ab.
- In Zeile 4 wird der Variablen `$name` `<STDIN>` zugewiesen. `STDIN` ist ein Handle, der Eingaben von der Shell liest, und zwar in unserem Beispiel eine Zeile. Da diese Eingabe mit der Enter-Taste abgeschlossen werden muß, was einem Zeilenumbruch gleicht, muß dieser nach dem Einlesen entfernt werden, was das `chomp()` in Zeile 5 erledigt.

## Mehr Kontrolle:

Code: *hello\_4.pl*

```
01: #! /usr/bin/perl
02:
03: print "Wie ist dein Name? ";
04: chomp($name = <STDIN>);
05:
06: if ($name eq "Martin" or $name eq 'root') {
07:     print "Hallo Martin, schoen, dich zu sehen!\n";
08: }
09: else {
10:     print "Hallo $name\n";
11: }
```

- Zeile 4: ist identisch mit den Zeilen 4 und 5 des letzten Beispiels, nur die Schreibweise ist etwas kompakter.
- Zeile 6: Zwei Zeichenketten kann man mit dem Operator `eq` vergleichen.
- Zeile 6: Mit `if` (Bedingung) `{ .... }` kann man Anweisungen bedingt ausführen.
- Zeile 6: mit `or` oder `and` kann man mehrere Bedingungen miteinander verknüpfen.
- Zeile 9: falls die Bedingung des `if`'s nicht wahr ist, wird der Inhalt des Blocks `{ }` nach dem `if` nicht ausgeführt, sondern, wenn vorhanden, der Code des `else`-Blocks

## Wie lautet das Geheimwort?

Code: *hello\_5.pl*

```
01: #! /usr/bin/perl
02:
03: # Konfiguration
04: $geheimWort = "Hufflepuff";
05:
06: # Programmcode
07: print "Wie ist dein Name? ";
08: chomp($name = <STDIN>);
09:
10: if ($name eq "harry") {
11:     print "Schwing dich auf den Besen!\n";
12: } # if
13:
14: else {
15:     print "Hallo $name\n";
16: }
```

```

17:     print "Wie lautet das geheime Wort? ";
18:     chomp($wort = <STDIN>);
19:
20:     while ($wort ne $geheimWort) {
21:         print "Leider falsch, aber versuch es nochmal\n";
22:
23:         print "Wie lautet das geheime Wort? ";
24:         chomp($wort = <STDIN>);
25:
26:     } # while
27:
28:     print "Bravo, du hast das Geheimwort gefunden. Tritt ein\n";
29:
30: } # else

```

- Zeilen 3,6,12,26,30: Mit dem Zeichen # kann man einen Kommentar einleiten, der bis zum Zeilenende geht.
- Zeile 20: Mit dem Operator **ne** vergleicht man zwei Zeichenketten, und es gibt wahr zurück, wenn die beiden Zeichenketten unterschiedlich sind. (**ne** steht für not equal, nicht gleich). **ne** ist also das Gegenteil von **eq**.
- Zeile 20: der Befehl **while** (Bedingung) { ... } wiederholt den Block so lange, wie die Bedingung wahr ist. In unserem Fall lautet die Bedingung solange das \$wort anders als das \$geheimWort ist, führe den Code in dem Block aus (frage das Geheimwort nochmal ab, bis es stimmt).

## Zahlen:

Code: *hello\_6.pl*

```

01: #! /usr/bin/perl
02:
03: # Konfiguration
04: $requiredAge = 25;
05:
06: # Programmcode
07: print ("Wie alt bist du? ");
08: chomp( $age = <STDIN> );
09:
10: if ($age == $requiredAge) {
11:     print "Ah, du bist genau $requiredAge. Glueck gehabt\n";
12: } # if
13:
14: elsif ($age < $requiredAge) {
15:     print "Sorry, du bist noch nicht " . $requiredAge . "\n";
16: } # elsif
17:
18: else {
19:     print "Ah, du bist alt genug, naemlich $age\n";
20: } # else

```

- Zeile 04 enthält nun eine Zahl, keine Zeichenkette
- Zeile 07/15: Man kann print auch als Funktion schreiben (mit Klammern), wenn man will, muß aber nicht (Zeilen 11/19)
- Zeile 14: es gibt nicht nur **if - else**, sondern man kann es auch noch erweitern: **if - elsif - else** (für C/Java-Programmierer: es heisst nicht elseif)

- Vergleichsoperatoren sind für Zahlen anders als für Zeichenketten

Zahl	Zeichenkette	Bedeutung
==	eq	Sind die beiden gleich?
!=	ne	Sind die beiden nicht gleich?
<	lt	Ist der linke Wert kleiner?
>	gt	Ist der rechte Wert kleiner?
<=	le	Ist der linke Wert kleiner oder gleich?
>=	ge	Ist der rechte Wert kleiner?
<=>	cmp	-1 wenn links kleiner, 0 wenn gleich, +1 wenn links größer

Achtung: = ist eine Zuweisung, kein Vergleich:  $\$x = \$y$

- Dualismus Zahl <-> Zeichenkette: was wie aufgefasst wird, ergibt sich aus dem Zusammenhang, und es wird hinter den Kulissen umgewandelt (d.h. eigentlich werden bei Bedarf beide Werte gespeichert, wenn nötig; man kann also auch Zahlen mit String-Vergleichen behandeln, oder umgekehrt, wenn man das wirklich will)

## Zahlenformate

Zahlen können auf viele verschiedene Arten angegeben werden, z.B.

Zahl	Beschreibung
123	"normale" Dezimalzahl
123_456_789	bessere Lesbarkeit für lange Zahlen (optional mit _)
0123	Oktalzahl (führende Ziffer 0)
0xFF	Hexadezimalzahl (führende Ziffer 0 und danach ein x)
0b1101_1100	Binär (führende Ziffer 0 und danach ein b) (_ für bessere Lesbarkeit)
6.02e23	Wissenschaftliche Notation
1.14159	Fließkommazahl (Punkt ist Trennzeichen)

## Rechenoperationen

Operation	Erklärung
$\$z = \$x + \$y$	Addition
$\$z = \$x - \$y$	Subtraktion
$\$z = \$x * \$y$	Multiplikation
$\$z = \$x / \$y$	Division (Ergebnis: wenn möglich, eine Integerzahl, sonst eine Fließkommazahl).
$\$z = \$x \% \$y$	Modulo (Rest der Integerdivision)
$\$z = \$x ** \$y$	Potenzierung

```
print 4 + 2 * 3; # 10, nicht 18: Punkt- vor Strichrechnung
```

```
print (4+2) * 3; # 18
```

Wenn der Vorrang nicht eindeutig ist, besser ein paar Klammern setzen.

## Kurzschreibweisen

Kurzschreibweise	Langschreibweise	Erklärung
<b>\$i++</b> oder <b>++\$i</b>	<b>\$i = \$i + 1</b>	\$i wird um eins erhöht
<b>\$i--</b> oder <b>--\$i</b>	<b>\$i = \$i - 1</b>	\$i wird um eins vermindert
<b>\$i += 2</b>	<b>\$i = \$i + 2</b>	\$i wird um zwei erhöht
<b>\$i -= 5</b>	<b>\$i = \$i - 5</b>	\$i wird um 5 vermindert
<b>\$i *= 2</b>	<b>\$i = \$i * 2</b>	\$i wird verdoppelt
<b>\$i .= 0</b>	<b>\$i = \$i . "0"</b>	An \$i wird eine 0 angehängt (Behandlung als Zeichenkette durch den Punkt!)

Wenn man `++$i` oder `$i++` als eigenständigen Ausdruck verwendet, sind die beiden vom Verhalten her identisch (ebenso `--$i` und `$i--`). Wenn es sich jedoch um eine Zuweisung oder ähnliches handelt, verhalten sich die beiden unterschiedlich:

`$i++` gib zuerst den Wert von `$i` zurück und erhöht dann `$i` um eins.  
`++$i` erhöht zuerst den Wert von `$i` und gibt dann den erhöhten Wert zurück.  
 Ebenso verhält sich `$i--` und `--$i`

```
$i = 5;
$j = $i++; # $j=5, $i=5+1=6
$k = ++$i; # $k = $i+1=7, $i=6+1=7
```

## Wahrheit oder Lüge?

In Perl gibt es weder `TRUE` noch `FALSE`, sondern diese Werte werden durch Zahlen oder Zeichenketten dargestellt (fast genauso wie in C):

Wert	Bedeutung
0 oder "0"	Falsch
Leere Zeichenkette ""	Falsch
Wert nicht initialisiert: undef(ined)	Falsch
Alles andere ist:	Wahr

## Verschiedene Zählerschleifen

### While-Schleife: kompliziert

Code: `loop_1.pl`

```
01: #! /usr/bin/perl
02:
03: $max = 5;
04:
05: $counter = 0;
06: while ($counter <= $max) {
07:     print "$counter\t";
08:
09:     $counter++;
10: } # while
11:
12: print "\n";
```

Output:

```
bash-2.05b$ perl 1_loop.pl
0      1      2      3      4      5
```

- Zeile 3: die Variable, die die obere Grenze enthält, wird initialisiert
- Zeile 5: die Zählvariable wird initialisiert
- Zeile 6: Die Schleifenbedingung: solange \$counter kleiner oder gleich **\$max** ist, führe den Block aus
- Zeile 7: Gebe den Wert aus, und danach einen Tabulator \t
- Zeile 9: Erhöhe den **\$counter** um eins

### for-Schleife im C-Stil: noch komplizierter

Code: loop\_2.pl

```
1: #!/usr/bin/perl
2:
3: $max = 5;
4:
5: for ($counter=0; $counter <= $max; $counter++) {
6:     print "$counter\t";
7: } # for
8:
9: print "\n";
```

Der Output ist identisch

### foreach-Schleife im Perl-Stil:

Code: loop\_3.pl

```
1: #!/usr/bin/perl
2:
3: $max = 5;
4:
5: foreach $counter (0..$max) {
6:     print "$counter\t";
7: } # foreach
8:
9: print "\n";
```

- Zeile 5: `foreach $Laufvariable ($start..$end) { ... }`. Anstelle von `foreach` kann man auch `for` schreiben, oder umgekehrt
- `$start..$end` erzeugt eine Liste. Nicht nur Zahlen können verwendet werden, sondern auch Buchstaben:

Code: loop\_4.pl

```
1: #!/usr/bin/perl
2:
3: foreach $char ('a'..'z') {
4:     print "$char ";
5: } # foreach
6:
7: print "\n";
```

Output:

```
bash-2.05b$ perl 4_loop.pl
a b c d e f g h i j k l m n o p q r s t u v w x y z
bash-2.05b$
```

Aus der letzten foreach-Schleife kann man auch die Liste herausziehen:

Code: *loop\_5.pl*

```
1: #! /usr/bin/perl
2:
3: @liste = ('a'..'z');
4:
5: foreach $char ( @liste ) {
6:     print "$char ";
7: } # foreach
8:
9: print "\n";
```

- Zeile 3: Die Liste @liste wird initialisiert
- Zeile 5: Über die Liste @liste wird iteriert

### Einfache Suchfunktionen:

- **\$pos = index(\$text, \$suchString)**; sucht nach dem Vorkommen von **\$suchString** in **\$text** und gibt die Position (als Zahl) aus, an der **\$suchString** gefunden wurde. Die Zählweise beginnt dabei (wie bei Listen) mit 0, also das erste Zeichen ist 0, das zweite 1, usw. Wird **\$suchString** nicht in **\$text** gefunden, wird die Zahl -1 zurückgegeben.
- **\$pos = index(\$text, \$suchString, \$startPosition)**: zusätzlich kann eine **\$startPosition** angegeben werden, d.h. ab welchem Zeichen gesucht werden soll. Das ist oft nützlich, wenn man mehrere Vorkommen von **\$suchString** finden will.
- **\$pos = rindex(\$text, \$suchString)**; wie **index(...)**, nur beginne die Suche von hinten
- Mächtigere Wege finden wir im Bereich der regulären Ausdrücke

### Nützliche Stringfunktionen:

- **\$str = substr(\$text, \$pos)**: gibt den Inhalt von **\$text** von der Position **\$pos** bis zum Ende von **\$text** zurück
- **\$str = substr(\$text, \$pos, \$laenge)**: gibt **\$laenge** Zeichen von **\$text** ab der Position **\$pos** zurück.
- Mächtigere Wege finden wir im Bereich der regulären Ausdrücke

```
$string = "Perl is great";
print substr($string, -5); # great
print substr($string, 8); # great
print substr($string, 0, 1); # P
print substr($string, 5, 2); # is
```

### Variablentyp eindimensionale Liste/Array:

- Eine Liste beginnt stets mit dem Zeichen @ (z.B. @liste)
- Auf einen Wert der Liste kann man über \$ vor und ein Index in [] nach dem Listennamen zugreifen, z.B. **\$liste[3]** (in Perl6 bleibt dann auch da das @-Zeichen: @liste[3])
- Den Index des letzten Elementes bekommt man durch den Wert von  **\$#liste** heraus
- Die Anzahl der Elemente bekommt man durch  **scalar(@liste)**;

Code: loop\_6.pl

```
01: #! /usr/bin/perl
02:
03: @liste = ('a'..'z');
04:
05: foreach $index (0..$#liste) {
06:     print "$index => $liste[$index]\t";
07: } # foreach
08:
09: print "\n\n";
10: print "Index des letzten Elementes: $#liste\n";
11: print "Anzahl Elemente: ", scalar(@liste), "\n";
```

- Zeile 5: Iteriere über eine Liste mit den Zahlen 0 bis zum letzten Index der Liste **@liste**
- Zeile 6: Gib den aktuellen Wert von **\$index** aus, danach => und danach den Wert des entsprechenden Elementes von **@liste**, und danach einen Tabulator **\t**
- Zeile 9: zwei Zeilenumbrüche
- Zeile 11: Da **scalar(...)** eine Funktion ist, kann sie nicht innerhalb einer Zeichenkette ausgeführt werden. Deshalb wurde die Zeichenkette unterbrochen.

Output:

```
F:\apacheweb\test_8085\html\codes>6_loop.pl
0 => a 1 => b 2 => c 3 => d 4 => e 5 => f 6 => g 7 => h 8 => i 9 => j
10 => k 11 => l 12 => m 13 => n 14 => o 15 => p 16 => q 17 => r 18 => s 19 => t
20 => u 21 => v 22 => w 23 => x 24 => y 25 => z

Index des letzten Elementes: 25
Anzahl Elemente: 26
F:\apacheweb\test_8085\html\codes>
```

### Initialisierung einer Liste:

1. Jedes Element einzeln zuweisen:

```
$liste[0] = 'A';
$liste[1] = 'B';
$liste[2] = 'C';
$liste[3] = 'D';
```

2. Mehrere Elemente auf einmal zuweisen:

```
@liste = ('A', 'B', 'C', 'D');
```

3. Einen Bereich zuweisen

```
@liste2 = ('A'..'D');
```

4. Komfortablere Schreibweise mit **Whitespace** als Trennzeichen

```
@liste3 = qw(A B C D);
```

Whitespace ist eine Zeichenklasse: Leerzeichen, Tabulator, Zeilenumbruch. Auch folgendes ist erlaubt:

```
@liste4 = qw(A B
              C D);
```

## Einfache Listenfunktionen:

Funktion:	Beschreibung:	Beispiel:
<code>\$str = join("sep", @list);</code>	Vereinigt die Elemente einer Liste zu einer Zeichenkette, indem sie mit dem Trennzeichen <b>sep</b> zusammengehängt werden.	<code>\$string = join("\n", @liste4);</code>
<code>@list = split(/sep/, \$str);</code>	Teilt eine Zeichenkette in eine Liste auf, wobei immer beim Trennzeichen <b>sep</b> aufgeteilt wird	<code>@liste = split(/\n/, \$string);</code>
<code>push(@list, \$e1, \$e2);</code>	Fügt ein neues Element (oder eine Liste) ans Ende von <code>@list</code>	<code>push(@list, \$#list+1, \$#list+2);</code>
<code>\$lastE = pop(@list);</code>	Entfernt das letzte Element von <code>@list</code> und gibt es zurück	<code>\$lastE = pop(@list);</code>
<code>\$firstE = shift(@list);</code>	Entfernt das erste Element von <code>@list</code> und gibt es zurück	<code>\$firstE = shift(@list);</code>
<code>unshift(@list, \$e1, \$e2);</code>	Fügt ein neues Element (oder eine Liste) an den Beginn von <code>@list</code>	<code>unshift(@list, -2, -1);</code>

Code: `array_1.pl`

```
01: #! /usr/bin/perl
02:
03: @liste = ('A'..'J');
04:
05: $string = join(" x ", @liste);
06: print "JOIN : $string\n";
07:
08: @liste2 = split(/ x /, $string);
09:
10: print "SPLIT: ";
11: foreach $buchstabe (@liste2) {
12:     print "$buchstabe ";
13: } # foreach
14:
15: print "\n";
```

Output:

```
F:\apacheweb\test_8085\html\codes>perl 1_array.pl
JOIN : A x B x C x D x E x F x G x H x I x J
SPLIT: A B C D E F G H I J
F:\apacheweb\test_8085\html\codes>
```

## Lesen von Textdateien:

Vorgehensweise:

1. Öffnen einer Datei
2. eine oder mehrere Zeilen lesen
3. Datei schließen

In Perl könnte das folgendermaßen aussehen:

1. `open` (FILEHANDLE, "datei.txt");
2. `foreach` \$line ( <FILEHANDLE> ) { `print` \$line; }
3. `close` (FILEHANDLE);

Was passiert, wenn es die Datei "datei.txt" nicht gibt oder sie nicht gelesen werden kann? Perl selbst versucht da einfach, stillschweigend weiterzumachen, als wäre alles ok. Da muß man sich dann als Programmierer selbst drum kümmern. Perl hilft einem dabei, indem `open` einen wahren Wert (meistens 1) zurückgibt, wenn es ok ist, und einen falschen Wert (`undef`) bei einem Fehler. Eine Fehlerbeschreibung steht in der Perl-Variable `$!`

Code: `file_1.pl`

```
01: #! /usr/bin/perl
02:
03: $datei = "gibtEsNicht.txt";
04:
05: $ergebnis = open(FH, $datei);
06:
07: if ($ergebnis) { # alles ok gegangen
08:
09:     print "ok\n";
10:     close (FH); # wieder schliessen
11: } # if
12:
13: else { # Fehler
14:     die "Error: couldn't open file '$datei': $!";
15: } # else
```

Output:

```
F:\apacheweb\test_8085\html\codes>perl 1_file.pl
Error: couldn't open file 'gibtEsNicht.txt': No such file or directory at 1_file
.pl line 14.
```

Ein Zeilenumbruch nach der Ausgabe des `die` unterdrückt das `at 1_file.pl line 14.`, z.B.

```
die "Error: couldn't open file '$datei': $!\n";
```

Esi hilft oft, wenn man den Dateinamen in einer Variable abspeichert, und diese Variable dann sowohl beim `open` als auch bei der Fehlermeldung ausgibt. Wenn man den Namen nämlich in der Fehlermeldung hardcoded, dann passt oft die Fehlermeldung nicht zur Ursache, und man sucht den Fehler oft an der falschen Stelle.

Es gibt jedoch noch schönere Möglichkeiten zur Fehlerabfrage:

Code: `file_2.pl`

```
01: #! /usr/bin/perl
02:
03: $datei = "gibtEsNicht.txt";
04:
05: if( open(FH, $datei) ) {
06:     print "ok\n";
07:     close (FH); # wieder schliessen
08: } # if
09:
10: else { # Fehler
11:     die "Error: couldn't open file '$datei': $!\n";
12: } # else
```

Code: file\_3.pl

```
01: #! /usr/bin/perl
02:
03: $datei = "gibtEsNicht.txt";
04:
05: unless( open(FH, $datei) ) {
06:     die "Error: couldn't open file '$datei': $!\n";
07: } # if
08:
09: else { # ok
10:     print "ok\n";
11:     close (FH); # wieder schliessen
12: } # else
```

- **unless (...)** bedeutet **if (not ...)** oder **if (! ...)**
- Da durch das **die** das Programm beendet wird, könnte man auf das **else** auch verzichten

Code: file\_4.pl

```
1: #! /usr/bin/perl
2:
3: $datei = "gibtEsNicht.txt";
4:
5: open(FH, $datei) or die "Error: couldn't open file '$datei': $!\n";
6:
7: print "ok\n";
8: close (FH); # wieder schliessen
9:
```

- In Perl gibt es (wie auch in C/C++, Java, ...) eine Kurzschlußauswertung bei booleschen Ausdrücken, d.h. wenn ein Ausdruck nicht mehr wahr werden kann, wird er auch nicht mehr ausgewertet. Das **or** ist ein logisches ODER, d.h. der Gesamtausdruck ist wahr, wenn eine der beiden Seiten wahr ist (oder auch beide) und falsch, wenn beide Seiten falsch sind. Dieses Verhalten wird in Perl häufig dazu mißbraucht, ein paar Buchstaben Code zu sparen. Der Ausdruck **Aktion or die ...** wird sehr häufig verwendet, um eine Aktion auf Fehler zu überprüfen.

*Anmerkung: Falls der erste Ausdruck falsch ist, wird der zweite ausgewertet und zurückgegeben.*

```
print 0 or 4; => 4
```

```
print 2 or 4; => 2
```

- Das logische UND heißt **and** und funktioniert analog, nur daß da beide Seiten wahr sein müssen, um als Ergebnis wahr zurückzuliefern.  

```
print 0 and 4; => 0
```

```
print 2 and 4; => 2
```
- **and** und **or** werden häufig auch als Bedingungen verwendet, z.B.  

```
if ($x > 10 and $x < 20) { Anweisung; }
```

```
if ($x < 10 or $x > 20) { Anweisung; }
```
- Für C-Programmierer gibt es auch noch **||** und **&&**, allerdings mit höherer Bindung als **or** und **and**  

```
open (FH, $filename) || die;
```

```
($value > 2) && print "Der Wert $value ist groesser als 2\n";
```
- Ich empfehle, im Zweifel **and** und **or** zu verwenden (eventuell mit Klammern), weil die einfach besser lesbar sind.

## Datei einlesen und am Bildschirm ausgeben:

Code: *file\_5.pl*

```
01: #! /usr/bin/perl
02:
03: $datei = "file_5.pl";
04:
05: unless( open(FH, $datei) ) {
06:     die "Error: couldn't open file '$datei': $!\n";
07: } # if
08:
09: foreach $line (<FH>) {
10:     print $line;
11: } # foreach
12:
13: close (FH);
14:
```

- Zeile 9: **foreach \$line (<FH>)** { liest die Datei auf einmal ein:

Code: *file\_6.pl*

```
01: #! /usr/bin/perl
02:
03: $datei = "file_6.pl";
04:
05: unless( open(FH, $datei) ) {
06:     die "Error: couldn't open file '$datei': $!\n";
07: } # if
08:
09: print foreach <FH>;
10:
11: close (FH);
12:
```

- Zeile 9: dasselbe wie oben, nur eine Kurzschreibweise. Da **print** eine Liste erwartet, könnte man das **foreach** auch weglassen: **print <FH>;**

Code: *file\_7.pl*

```
01: #! /usr/bin/perl
02:
03: $datei = "file_7.pl";
04:
05: unless( open(FH, $datei) ) {
06:     die "Error: couldn't open file '$datei': $!\n";
07: } # if
08:
09: while ($line =<FH>) {
10:     print $line;
11: } # foreach
12:
13: close (FH);
14:
```

- Zeile 9: dasselbe mit einer **while**-Schleife: Da **while** jedoch nicht wie **for(each)** eine Liste aufnimmt, sondern einen skalaren Ausdruck, wird die Datei Zeile für Zeile eingelesen und beim nächsten Zeilenbeginn durch die nächste Zeile ersetzt. Dies ist vorteilhaft, wenn man große Dateien hat, weil da immer nur eine Zeile im Speicher landet, während bei **for(each)** die komplette Datei in den Speicher geladen wird.
- **<FH>** gibt also, je nach Kontext, das Ergebnis in unterschiedlicher Weise zurück (siehe die beiden folgenden Beispiele):

- o Skalarer Kontext: gibt eine Zeile der Datei zurück
- o Listenkontext: gibt die komplette Datei als Liste zurück
- Da **while** die Schleife solange ausführt, solange die Schleifenbedingung wahr ist, wird die komplette Datei eingelesen.
- Was passiert, wenn in der letzten Zeile der Datei eine 0 ohne einen folgenden Zeilenumbruch steht? Bei neueren Perl-Versionen wird da 0 ordentlich verarbeitet. Manche ältere Perl-Versionen werten da die 0 als Falsch aus, und brechen die Schleife ab, ohne daß die 0 noch behandelt wird. Deshalb ist es sicherer, da folgende Schreibweise zu verwenden:

Code: *file\_8.pl*

```
01: #! /usr/bin/perl
02:
03: $datei = "file_8.pl";
04:
05: unless( open(FH, $datei) ) {
06:     die "Error: couldn't open file '$datei': $!\n";
07: } # if
08:
09: while (defined($line =<FH>)) {
10:     print $line;
11: } # foreach
12:
13: close (FH);
14:
```

- Zeile 9: Die Funktion **defined(\$variable)** ist ein wenig strenger und überprüft, ob die Variable **\$variable** einen Wert enthält oder nicht. Wenn sie **undef** ist, gibt sie falsch zurück, sonst wahr.

Wert	Wahrheitswert	defined(Wert)
"abcde", 3	Wahr	Wahr
0, ""	Falsch	Wahr
undef	Falsch	Falsch

- **undef** bedeutet, die Variable hat keinen Wert (in anderen Sprachen: NUL, NULL, NULL, ...)
- Man kann auch einer skalaren Variablen den Wert **undef** zuweisen: **\$line = undef;**
- **undef** kann man auch als Prozedur verwenden: **undef(@liste);** Das leert den Speicher von **@liste** und wirft sie aus dem Programm raus.
- Wenn man nur den Inhalt einer Liste leeren will, kann man schreiben: **@liste = ();** Das weist der **@liste** eine leere Liste zu.
- Eine Liste kann man auch leeren, indem man ihren letzten Index auf -1 setzt:  **\$#liste = -1;**

In den folgenden beiden Beispielen sieht man das unterschiedliche Verhalten von **while** und **foreach**:

Code: *file\_9.pl*

```
01: #! /usr/bin/perl
02:
03: print "Eingabe: ";
04: while (defined($line = <STDIN>)) {
05:     chomp($line);
06:     print "Input: $line\n";
07:     if ($line eq 'END') {
08:         last;
09:     } # if
10:     print "Eingabe: ";
11: } # while
12:
```

- Dieses Beispiel wirkt sehr interaktiv, weil man eine Zeile eingibt und sofort eine Antwort herausbekommt
- Man kann in der Schleife überprüfen, ob ein Ende der Eingaben gewünscht ist (hier durch die Eingabe der Zeichenkette END).

Code: *file\_10.pl*

```
1: #! /usr/bin/perl
2:
3: foreach $line (<STDIN>) {
4:     chomp($line);
5:     print "Input: $line\n";
6: } # while
7:
```

- In diesem Beispiel werden im Schleifenkopf alle Eingaben eingelesen, und erst dann beginnt die Schleife zu laufen und gibt alle Zeilen aus. Da der Einlesevorgang nicht von der Schleife, sondern von <STDIN> gesteuert wird, muß man den Einlesevorgang abbrechen, indem man mit der Tastenkombination <Strg><d> (oder unter Windows auch: <Strg><z>) ein EndOfFile signalisiert. Ein Überprüfen auf die Eingabe von END wäre in diesem Beispiel nicht besonders sinnvoll, weil sie einfach zu spät erfolgen würde, nämlich nach der Eingabe der gesamten Liste.

Bei der While-Schleife gibt es noch folgende Kurzschreibweise:

```
while (defined($_ = <FH>)) {
    print $_;
} # while
```

\$\_ ist eine eingebaute Perl-Variable, die häufig die Standardvariable von Funktionen ist. Sie kann man häufig auch weglassen, z.B

```
while (<FH>) { # hier wird automatisch an $_ zugewiesen und auf defined überprüft
    print; # hier wird automatisch die Standardvariable $_ benutzt
} # while
```

Und weil das Motto von Perl TIMTOWDI (=There Is More Than One Way to Do It) lautet, funktioniert auch

```
print while <FH>;
```

Man muß halt von Fall zu Fall unterscheiden, welches der Konstrukte die gewählte Absicht am besten ausdrückt.

### IO- und File-Handles:

Es gibt in Perl sogenannte Handles, über die man auf Eingaben zugreifen und Ausgaben senden kann. Ein Handle ist ein Variablentyp, der auf eine Datei, auf die Konsole oder auch auf Netzwerksockets usw. zeigen kann. Standardmäßig hat ein Perl-Script (wie fast jedes andere Programm) drei sogenannter IO-Handles verbunden:

- **STDIN**: darüber kann man Eingaben über die Konsole einlesen (oder auch via Pipes), vergleiche <STDIN>
- **STDOUT**: darüber kann man Ausgaben in die Konsole machen, z.B. mit **print**
- **STDERR**: darüber kann man Fehler und Warnungen in die Konsole ausgeben, z.B. **die** oder **warn**

Diese drei IO-Handles sind standardmäßig mit Perl verbunden, d.h. man muß sie nicht öffnen, sondern das erledigt Perl für uns.

Man kann mit **open (FILEHANDLE, \$dateiname)** auch eine Datei zu einem Handle verbinden und dann darüber ansprechen. Diese Handles nennt man Filehandles. Perl stellt identische Funktionen für IO-Handles oder Filehandles zur Verfügung (z.B. <HANDLE>, **print HANDLE \$text**). Da **STDOUT** der Standardhandle für print usw. ist, kann man den auch weglassen, und braucht nicht immer zu schreiben: **print STDOUT \$text**; sondern kann die Kurzschreibweise verwenden, wie wir es bisher getan haben: **print \$text**;

## Dateien schreiben:

Code: *file\_write\_1.pl*

```
01: #! /usr/bin/perl
02:
03: $datei = "file_write_1.txt";
04:
05: unless( open(FH, "> $datei") ) {
06:     die "Error: couldn't write to file '$datei': $!\n";
07: } # if
08:
09: foreach $index (0..10) {
10:     print (FH "Dies ist Zeile $index\n");
11: } # foreach
12:
13: close (FH);
14:
```

- Zeile 05: öffne die Datei **\$datei** unter dem Filehandle **FH** zum schreiben. **open (FH, "> \$datei")**. Man kann auch schreiben: **open (FH, ">" . \$datei)** oder: **open (FH, '>' . \$datei)**.
- **Achtung: Wenn da eine schon vorhandene Datei zum Schreiben geöffnet wird, wird die überschrieben.**
- Zeile 10: schreibe einen Text in die Datei, die mit dem Filehandle **FH** zum Schreiben geöffnet ist.
- **Achtung: zwischen FH und der Zeichenkette darf kein Komma angegeben werden!**
- Anmerkung: Man kann Dateien auch folgendermaßen zum Lesen öffnen: **open (FH, "< \$datei")**. Wir haben bisher jedoch die abgekürzte Version **open (FH, \$datei)** verwendet.

Code: *file\_write\_2.pl*

```
01: #! /usr/bin/perl
02:
03: $dateiZumLesen      = "file_write_2.pl";
04: $dateiZumSchreiben = "file_write_2.txt";
05:
06: unless( open (IN, $dateiZumLesen) ) {
07:     die "Error: couldn't read from file '$dateiZumLesen': $!\n";
08: } # unless
09:
10: unless( open(OUT, "> $dateiZumSchreiben") ) {
11:     die "Error: couldn't write to file '$dateiZumSchreiben': $!\n";
12: } # if
13:
14: foreach $line (<IN>) {
15:     chomp($line);
16:     $output = reverse($line);
17:     print (OUT "$output\n");
18: } # foreach
19:
20: close (IN);
21: close (OUT) or
22:     die "Fehler beim Schliessen von '$dateiZumSchreiben': $!\n";
23:
24:
```

- Zeile 6-8: Datei zum Lesen öffnen
- Zeile 10-12: Datei zum Schreiben öffnen
- Zeile 20: Lesedatei wieder schließen
- Zeile 21: Schreibdatei mit Fehlerauswertung wieder schließen (z.B. Festplatte voll)

- Zeile 16: **reverse()** dreht eine Zeichenkette (Zeichen für Zeichen) oder eine Liste (Element für Element) um. **reverse()** reagiert jedoch unterschiedlich, je nachdem, ob man es in skalarem Kontext oder im Listenkontext ausführt. Da wir es durch eine Zuweisung an eine skalare Variable (=S**output**) im skalaren Kontext ausführen, wird die Zeichenkette Zeichen für Zeichen umgedreht und in Zeile 17 ausgegeben. Wenn man Zeile 16 und 17 zusammenziehen will, kann man deshalb nicht schreiben  
**print (OUT reverse(\$line), "\n");**  
weil print einen Listenkontext erzwingt, **\$line** dann als Liste mit einem Element angesehen wird und deshalb scheinbar nichts passiert. Durch den richtigen Operator kann man jedoch das **reverse()** in einen skalaren Kontext zwingen, z.B.  
**print (OUT reverse(\$line) . "\n");**  
**print (OUT scalar(reverse(\$line)), "\n");**  
und die Zeichenkette wird - wie gewünscht - Zeichen für Zeichen umgedreht in die Datei, die mit dem Filehandle OUT verbunden ist, geschrieben..

### Ausgabefunktionen:

- **print** (\$zeichenkette);
- **print** (FH \$zeichenkette);
- **printf** ("Muster", @werte);
- **printf** (FH "Muster", @werte);

Die Klammern sind hier optional.

Die Funktion **printf()** wurde aus den Sprachen C/C++ übernommen und funktioniert (fast) identisch:

z.B. `printf("Zahl: %i Zeichenkette: %s \n", 20, 'text');`

Mit % werden Platzhalter (eventuell mit Formatanweisungen) angegeben. Dabei sind folgende möglich:

Platzhalter:	Ersetzung:	Beispiel:	Ausgabe:
%%	ein normales %-Zeichen	<code>printf("%%");</code>	%
%s	eine Zeichenkette	<code>printf("%s", 'abc');</code>	abc
%d oder %i	eine Integerzahl	<code>printf("%i", 255);</code>	255
%o	Integer in Oktalschreibweise	<code>printf("%o", 255);</code>	377
%x	Integer in Hexadezimalschreibweise	<code>printf("%x", 255);</code>	ff
%f	eine Fließkommazahl	<code>printf("%f", 30.22);</code>	30.220000
%e	Fließkommazahl in wissenschaftlicher Notation	<code>printf ("%e", 30,22);</code>	3.022000e+001

Es gibt noch weitere Platzhalter, die jedoch nur selten verwendet werden.

Man kann bei einem Platzhalter noch weitere Optionen angeben, und zwar zwischen dem % und dem Buchstaben:

Zeichen:	Bedeutung:	Beispiel:	Ausgabe:
Zahl	Die minimale Feldlänge	<code>printf('%10s %4s', 'ab', 'cd');</code>	ab cd
Leerzeichen	Bei Rechtsbündiger Ausgabe werden vor die Zahl/Zeichenkette Leerzeichen ausgegeben	<code>printf('% 20s', 'abcd');</code>	abcd
Ziffer 0	Bei Rechtsbündiger Ausgabe werden vor die Zahl/Zeichenkette Nullen geschrieben	<code>printf('%04i %04i', 20, 3);</code>	0020 0003
+	Stelle vor positiven Zahlen ein Plus dar	<code>printf('%+4i', 20);</code>	+20
-	Gib die Zeichenkette/Zahl linksbündig aus (Standard ist: Rechtsbündig)	<code>printf('%-+4i', 20);</code>	+20
Zahl.Zahl	Genauigkeit bei Fließkommazahlen (kann man wunderbar zum Runden verwenden).	<code>printf('%4.2f', 30.2272);</code>	30.23

Damit kann man die Ausgabe recht schön formatieren

Damit man diese Formate auch bei Zuweisungen verwenden kann, gibt es in Perl zusätzlich noch die Funktion **sprintf()**, die identisch wie **printf** funktioniert, allerdings die Zeichenkette einer Variablen zuweist, z.B.

```
$output = sprintf("%4d %10s\n", $lineNumber, "Zeile" . $lineNumber);
$gerundet = sprintf("%4.3f", 12.1259);
```

### Globale und lokale Variablen:

Bisher haben wir immer globale Variablen verwendet. Es gibt jedoch in Perl wie auch in den meisten anderen Sprachen die Möglichkeit, lokale Variablen mit eingeschränktem Gültigkeitsbereich zu verwenden. Diese haben den Vorteil, daß man sich keine Gedanken zu machen braucht, was außerhalb eines Bereiches mit einer anderen Variable desselben Namens passiert.

In Perl gibt es das Konzept der Blöcke, z.B. **foreach (...)** { **BLOCK** } oder **while (...)** { **BLOCK** } oder auch, wie wir bald bei Subroutinen sehen werden, **sub Name** { **BLOCK** }. Man kann in so einen Block Variablen deklarieren, die nur innerhalb dieses Blockes gültig sind. Das Schlüsselwort, um eine solche Variable zu deklarieren, heißt **my**, z.B. **my \$text**; Während der Deklaration kann man auch gleich einen Wert zuweisen und die Variable so initialisieren. Ohne dieser Intitalisierung hat eine Variable den Wert **undef**.

Code: *my\_1.pl*

```
01: #! /usr/bin/perl
02:
03: my $var = 5;
04:
05: print "Vor dem Block: $var\n";
06:
07: if ($var == 5) {
08:     my $var = 20; # ist andere Variable als oben
09:     print "Im Block: $var\n";
10: } # if
11:
12: print "Nach dem Block: $var\n";
```

```
F:\apacheweb\test_8085\html\codes>perl my_1.pl
Vor dem Block: 5
Im Block: 20
Nach dem Block: 5

F:\apacheweb\test_8085\html\codes>
```

Variablen (in Perl mit **my**) zu deklarieren ist fast jeder Programmiersprache ein sehr guter Stil, so auch in Perl, weil man die Programme in kleine logisch-vollständige Blöcke unterteilen kann, die von einander unabhängig sind. Man kann Perl sagen, es solle einen dazu zwingen, alle Variablen vor der (oder bei der ersten) Verwendung zu deklarieren, um so schwierig zu findende Tippfehler einfacher unter Kontrolle bekommen.

### Die Zähmung des Kamels:

wo liegt der Fehler bei folgendem Code?

Code: *my\_2.pl*

```
1: #! /usr/bin/perl
2:
3: foreach my $variableOhneNamen (1..3) {
4:     print "$variableOhneNamen\n";
5: } # foreach
```

```
F:\apacheweb\test_8085\html\codes>perl my_2.pl

F:\apacheweb\test_8085\html\codes>
```

Wenn man die Compileroption `strict` verwendet, wird der Fehler schnell klarer:

Code: *my\_3.pl*

```
1: #! /usr/bin/perl
2: use strict;
3:
4: foreach my $variableOhneNamen (1..3) {
5:     print "$variableOhneNamen\n";
6: } # foreach
```

Output:

```
F:\apacheweb\test_8085\html\codes>my_2.pl
Global symbol "$variableOhneNamen" requires explicit package name at F:\apacheweb\test_8085\html\codes\my_2.pl line 5.
Execution of F:\apacheweb\test_8085\html\codes\my_2.pl aborted due to compilation errors.

F:\apacheweb\test_8085\html\codes>
```

Diese Meldung sagt, daß die Variable **\$variableOhneNamen** in der Zeile 5 nicht mit `my` deklariert worden ist. Wir haben sie jedoch in der Zeile 4 deklariert und initialisiert. Also muß da ein Rechtschreibfehler vorliegen. Und wenn man genau hinsieht, bemerkt man, daß ich anstelle von dem Buchstaben `O` die Ziffer `0` geschrieben habe. Bei größeren Programmen kann die Suche nach einem solchen Fehler oft Stunden dauern, wenn man `use strict;` nicht verwendet.

Code: *my\_4.pl*

```
1: #!/usr/bin/perl
2:
3: my $var = "abcdefg\n";
4: print "abcdefg", var, "\n";
```

Output:

```
F:\apacheweb\test_8085\html\codes>my_4.pl
abcdefgvar
F:\apacheweb\test_8085\html\codes>
```

Code: *my\_5.pl*

```
1: #!/usr/bin/perl
2: use strict;
3:
4: my $var = "abcdefg\n";
5: print "abcdefg", var, "\n";
```

Output:

```
F:\apacheweb\test_8085\html\codes>my_5.pl
Bareword "var" not allowed while "strict subs" in use at F:\apacheweb\test_8085\
html\codes\my_5.pl line 5.
Execution of F:\apacheweb\test_8085\html\codes\my_5.pl aborted due to compilatio
n errors.
F:\apacheweb\test_8085\html\codes>
```

Dies sagt, daß in der Zeile 5 ein **\$** vor **var** vergessen wurde.

**use strict;** achtet auch noch auf weitere Themen, die jedoch erst im Fortgeschrittenenkurs behandelt werden.

Es gibt noch eine Hilfe, die versucht, einen auf verdächtige Konstrukte hinzuweisen, z.B. **use warnings;** (ab Perl5.6):

Code: *my\_6.pl*

```
1: #!/usr/bin/perl
2: use strict;
3: use warnings;
4:
5: my $var;
6: unless ($var) {
7:     my $var = 20;
8: } # unless
9: print "Variable: $var\n";
```

Output:

```
F:\apacheweb\test_8085\html\codes>my_6.pl
Use of uninitialized value in concatenation (.) or string at F:\apacheweb\test_8085\html\codes\my_6.pl line 9.
Variable:
F:\apacheweb\test_8085\html\codes>
```

Perl ist, wie wir nach und nach kennenlernen werden, eine äußerst mächtige Programmiersprache. Sie versucht zu ahnen, was der Benutzer will, und versucht das dann zu tun. Diese Verhaltensweise ist jedoch oft gefährlich, weil man viele Fehler oft nur sehr schwer entdeckt.

Die beiden Ausdrücke **use strict;** und **use warnings;** helfen einem dabei, Perl so unter Kontrolle zu bekommen, daß es einem bei der Fehlersuche hilft und Perl zu einer vernünftigen, vollwertigen Programmiersprache macht. In verschiedenen Webforen wie z.B. <http://www.perl-community.de/> oder <http://www.perlmonks.org/> wird häufig Fragen zu Problemen gestellt, die der Programmierer mit der Verwendung von **strict** und **warnings** viel schneller selbst hätte lösen können. Und auch guten Programmierern passiert es gelegentlich, daß sie vor einem unerklärlichen Fehler stehen und dann um Hilfe suchen, nur weil sie mal schnell was ohne **strict** und **warnings** geschrieben haben. Aus diesem Grund verwende ich für jedes Programm, das länger als so 2-3 Zeilen ist, automatisch **strict** und **warnings**, und irgendwie habe ich fast nie unerklärliche Fehler in meinen Codes. Um die Verwendung von **strict** und **warnings** zu demonstrieren, werde ich ab sofort alle Codebeispiele mit **strict** und **warnings** angeben.

Die Variable `$_` kann man allerdings nicht mit `my` deklarieren, weil es sich um eine eingebaute Perl-(Package-) Variable handelt. Wenn man Gültigkeit des Wertes von `$_` auf einen Block beschränken will, muß man das altbackene `local` verwenden, was eine globale Variable mit demselben Namen durch eine lexikalische (= nur in diesem Bereich gültige) überdeckt. Die globale Variable wird dadurch nicht verändert; sie ist lediglich im aktuellen Bereich dann nicht mehr sichtbar.

```
while (defined( local($_) = <FH>)) {
    print $_;
} # while
```

Eine `while`-Schleife lokalisiert das `$_` nicht automatisch, während eine `for`-Schleife dies schon macht.

Bei neueren Perl-Versionen reicht es auch zu schreiben:

```
while (local $_ = <FH>) {
    print;
} # while
```

### Ausführen von externen Programmen:

In Perl gibt es mehrere Möglichkeiten, externe Kommandos auszuführen. Dabei muß man sich entscheiden, was man vom externen Programm wissen will.

Wenn man herausfinden will, ob ein externes Programm korrekt gelaufen ist (mit `exitcode` 0), aber der Output interessiert einen nicht, dann ist der Befehl **system()** angebracht:

Code: *command\_1.pl*

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: unless (system("ls -la") == 0) {
06:     die "Error: $?\n";
07: } # unless
08: else {
09:     print "All right\n";
10: } # else
```

Wenn man **system()** eine Zeichenkette mitgibt, wird (meistens) eine Shell geöffnet, die das Kommando interpretiert und dann ausführt. Wenn man dies vermeiden will, übergibt man system einfach eine Liste:

Code: *command\_2.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: unless (system("ls", "-la") == 0) {
06:     die "Error: $?\n";
07: } # unless
08: else {
09:     print "All right\n";
10: } # else
```

So wird die Shell nicht verwendet, und man kann überdies auch ein klein wenig Laufzeit sparen.

Wenn man an der Ausgabe eines Programmes interessiert ist, kann man sich zwischen den folgenden Möglichkeiten entscheiden:

1. **Backticks**: (sind auf der deutschen Tastatur rechts neben dem ?)

Code: *command\_3.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my @result = `dir`; # Rueckgabe als Liste, eine Zeile ist ein Element
06: die "Error 1: $!\n" if $!;
07: print @result, "\n\n";
08:
09: my $result = `dir`; # der gesamte Output steht in $result
10: die "Error 2: $!\n" if $!;
11: print $result;
```

Anstelle von Backticks kann man auch **my \$result = qx(ls -l);** schreiben

2. **Pipe-open**:

Dies funktioniert im Grunde genauso wie das lesen von einer Datei:

Code: *command\_4.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $command = 'ls -la';
06: unless(open (CMD, "$command |")) {
07:     die "Error in executing '$command': $!\n";
08: } # unless
09: while (<CMD>) {
10:     chomp($_);
11:     print "Zeile: $_\n";
12: } # while
13: close (CMD) or die "Error in closing '$command': $!\n";;
```

Ebenso kann man Daten an ein anderes Programm pipen (vorausgesetzt, das Programm versteht die Daten per Pipe):

Code: *command\_5.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $email = 'irgendwas@local';
06: my $sendmail = '/usr/bin/sendmail -t';
07:
08: unless (open (SENDMAIL, "| $sendmail")) {
09:     die "Error: couldn't open '$sendmail': $!\n";
10: }
11: print SENDMAIL "To: irgendetwem <$email>\n";
12: print SENDMAIL "From: sonstwem <$email>\n";
13: print SENDMAIL "Subject: Testmail\n";
14: print SENDMAIL "Email-Text\n";
15:
16: close (SENDMAIL) or die "Error in closing '$sendmail': $!\n";
17:
```

Es gibt noch weitere Möglichkeiten, externe Programme auszuführen und fernzusteuern; da diese jedoch alle Module benötigen, behandle ich sie hier nicht.

Bei der Ausführung externer Kommandos lege ich sehr viel Wert die Fehlerabfrage, weil sie mir schon Stunden von Fehlersuche erspart hat.

### Datenstrukturen:

Nochmal zurück zum Beispiel *hello\_5.pl*:

Code: *hello\_5.pl*

```
01: #! /usr/bin/perl
02:
03: # Konfiguration
04: $geheimWort = "Hufflepuff";
05:
06: # Programmcode
07: print "Wie ist dein Name? ";
08: chomp($name = <STDIN>);
09:
10: if ($name eq "harry") {
11:     print "Schwing dich auf den Besen!\n";
12: } # if
13:
14: else {
15:     print "Hallo $name\n";
16:
17:     print "Wie lautet das geheime Wort? ";
18:     chomp($wort = <STDIN>);
19:
20:     while ($wort ne $geheimWort) {
21:         print "Leider falsch, aber versuch es nochmal\n";
22:
23:         print "Wie lautet das geheime Wort? ";
24:         chomp($wort = <STDIN>);
25:
26:     } # while
27:
28:     print "Bravo, du hast das Geheimwort gefunden. Tritt ein\n";
29:
30: } # else
```

Wie kann man diesen Zutrittsmechanismus so verfeinern, daß man für jede Person einen eigenen Benutzernamen

und ein eigenes Passwort vergeben kann?

1. Die Namen und Benutzernamen in eine Liste packen, und zwar jeweils mit einem Trennzeichen getrennt (z.B. #)

```
my @accounts = ("dumbledore#geheim", "ron#weasley", "harry#potter", "neville#");
```

und dann \$name und \$wort eingeben lassen, und dann über die Liste @accounts iterieren:

```
for my $combination (@accounts) {
    my ($sName, $sPassword) = split(/#/ , $combination, 2);
    if ($name eq $sName and $wort eq $sPassword) {
        print "ok\n";
    } # if
} # for
```

oder den umgekehrten Weg wählen:

```
for my $combination (@accounts) {
    my $test = join("#", $name, $wort);
    if ($combination eq $test) {
        print "ok\n";
    } # if
} # for
```

Diese Vorgehensweisen sind gut, aber es gibt noch bessere, indem man eine andere Datenstruktur verwendet:

**Hashes:**

Hashes (oft auch als assoziative Arrays bezeichnet) sind eine Datenstruktur wie ein Array, nur daß auf einzelne Elemente über eine Zeichenkette zugegriffen werden kann anstelle von einer Zahl. Den Index nennt man da Schlüssel (oder englisch: key) und den Wert einfach Wert (oder englisch: value). Man kann einfach von einem Key auf einen Value zugreifen, umgekehrt wird es aber komplizierter. Genau wie bei einer Liste kann ein Hashkey auf den Inhalt einer skalaren Variable zeigen (derzeit Zeichenketten oder Zahlen; im Fortgeschrittenenkurs werden auch Referenzen auf andere Daten(strukturen) behandelt).

Beispiel Array:	Array	Hash	Beispiel Hash:
\$array[3] = 20;	Index ist Zahl	Index ist String	\$hash{'zwanzig'} = 20;
[ ]	Eckige Klammern	Geschwungene Klammern	{ }
@liste	@ für ganze Liste	% für ganzen Hash	%hash

**@array**

Index:	Wert:
0	'abcde'
1	42
2	3.1415926354
3	\$skalareVariable
4	\$otherArray[2]
5	\$hash{'Antwort auf alle Fragen'}
...	...

**%hash**

Index:	Wert:
'code'	'abcde'
'Antwort auf alle Fragen'	42
'PI'	3.1415926354
'irgendwas'	\$skalareVariable
'arrayWert'	\$array[2]
'hashwert'	\$otherHash{'code'}
...	...

Wenn man einem Element eines Arrays oder Hashes ein Element eines anderen Arrays oder Hashes zuweist, wird der Wert kopiert. Die beiden Container werden dadurch nicht in irgendeiner Weise verbunden. Sogas könnte man mit Referenzen machen, wie wir später sehen werden.

Bei einem Hash kann man auf die Anführungszeichen beim Hashkey verzichten, wenn er keine Leerzeichen enthält,

nicht mit einem mathematischen (+, -) oder anderswertig reserviertem Zeichen (\$, @) beginnt, z.B. \$hash{abcd}, aber \$hash{'mit Leerzeichen'}

Code: hash\_01.pl

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my %passwords = (
06:     "dumbledore" => 'geheim',
07:     'ron'         => "weasly",
08:     "harry"       => "potter",
09:     "neville"    => '',
10: );
11:
12: print "Usernamen eingeben: ";
13: my $username = <STDIN>; chomp($username);
14:
15: print "Passwort eingeben: ";
16: my $password = <STDIN>; chomp($password);
17:
18: if (exists $passwords{$username} ) {
19:
20:     if ( $password eq $passwords{$username} ) {
21:         print "Passwort korrekt. Herzlich willkommen, $username\n";
22:     } # if
23:
24:     else {
25:         print "Passwort $passwords{$username} ist nicht korrekt\n";
26:     } # else
27:
28: } # if
29:
30: else {
31:     print "Kenne niemanden mit dem Usernamen $username\n";
32: } # else
33:
```

- Zeile 5-9: der Hash wird deklariert und gleich mit Keys und Values versehen
- Zeile 6-9: der => Operator ist eigentlich ein normales Komma, er wird jedoch häufig verwendet, weil er die Zusammengehörigkeit von Value zu Key besser veranschaulicht und den Wert davor "stringifiziert", das heisst, man kann die Anführungszeichen beim Key weglassen.
- Zeile 17: **exists \$hash{Key}** gibt wahr zurück, wenn der Schlüssel existiert, sonst falsch. Damit überprüfen wir, ob ein Benutzername überhaupt existiert.

### Initialisierung eines Hashes:

Code: hash\_02.pl

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my %hash1 = (
06:     "dumbledore" => 'geheim',
07:     'ron'         => "weasly",
08:     "harry"       => "potter",
09:     "neville"    => '',
10: );
11:
12: my %hash2 = (
13:     "dumbledore", 'geheim',
14:     'ron'         , "weasly",
15:     "harry"       , "potter",
```

```

16:         "neville" , '',
17:     );
18:
19: my %hash3 = qw(
20:     dumbledore geheim
21:     ron          weasly
22:     harry        potter
23:     neville     vergessen
24: );
25:
26: my %hash4 = ();
27: $hash4{dumbledore} = 'geheim';
28: $hash4{ron} = 'weasly';
29: $hash4{harry} = 'potter';
30: $hash4{neville} = '';

```

## Über einen Hash iterieren:

### 1. for(each)-Schleife

```

foreach my $key (keys %hash) {
    print "$key: $hash{$key}\n";
} # foreach

```

**keys( %hash )** ist eine Funktion, die die ganzen Schlüssel eines Hashes in einer Liste zurückgibt. Analog dazu gibt es noch **values( %hash )**, das die ganzen Werte zurückgibt. values wird aber sehr selten verwendet.

### 2. while-Schleife

```

while (my ($key, $value) = each %hash) {
    print "$key: $value\n";
} # while

```

Wobei da meistens die foreach-Schleife verwendet wird, weil man da noch einfach weitere Angaben machen kann, wie z.B. Sortierungen:

```

foreach my $key (sort( keys %hash)) {
    print "$key => $hash{$key}\n";
} # foreach

```

sort nimmt eine Liste entgegen (von keys), sortiert die dann Ascii-betisch und gibt die sortierte Liste wieder zurück. Dazu später genaueres.

- Bei Hashes ist der Key stets eindeutig und kann nur einen Wert aufnehmen. Wenn einem Hash ein Key/Value-Paar zugewiesen wird, dessen Key schon existiert, wird der vorherige Value einfach überschrieben.
- Mit **exists( \$hash{Element} )** kann man überprüfen, ob ein Key schon existiert.
- Mit **defined( \$hash{Element} )** kann man überprüfen, ob für einen Key schon ein Value existiert. Ein Key in einem Hash kann ohne Probleme den Wert **undef** haben, genauso wie ein Array (dies würde man dann "Array mit Löchern" nennen).
- Mit **keys(%hash)** bekommt man eine Liste von Hashkeys zurück.
- Mit **values(%hash)** bekommt man eine Liste von Werten zurück.
- Die Reihenfolge, in der die Keys und Values zurückgeliefert werden, ist nicht vorhersagbar; festgelegt ist lediglich, dass bei keys und bei values dieselbe Reihenfolge der Elemente zurückgegeben wird. Diese Reihenfolge kann sich jedoch von Perl-Version zu Perl-Version ändern, also sollte man sich nicht auf eine fixe Reihenfolge verlassen.
- Ein Hash ist in der Regel langsamer als ein Array, was jedoch häufig durch die einfachere Zugriffsweise auf einzelne Werte wieder wettgemacht wird.
- Der Hash ist eine der wichtigsten Datenstrukturen in Perl und häufig die Basis von komplexeren Datenstrukturen und Objekten.

- Hashes sind häufig etwas komplizierter zu verstehen, weil sie eine andere algorithmische Denkweise als Arrays erfordern.
- Ein Element eines Hashes kann man mit dem Befehl **delete** löschen, z.B. **delete \$hash{Element};**

### Wörter zählen mit Hashes:

Wir haben in einer Textdatei (wordlist.txt) alle Wörter dieses Vortrages, die länger als 8 Zeichen sind, und zwar ein Wort pro Zeile. Wie können wir herausfinden, welche Wörter öfter als 9 Mal vorkommen? Mit Listen wäre dies ein recht aufwendiges Unterfangen, mit einem Hash ist dies jedoch recht einfach lösbar:

Code: hash\_04.pl

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $textfile = 'wordlist.txt';
06:
07: unless (open (WORDS, $textfile)) {
08:     die "Error: can't open '$textfile': $!\n";
09: } # unless
10:
11: my %wordsCount = ();
12:
13: while (defined (my $word = <WORDS>)) {
14:     chomp($word);
15:
16:     if (exists $wordsCount{$word}) {
17:         $wordsCount{$word} ++;
18:     } # if
19:     else {
20:         $wordsCount{$word} = 1;
21:     } # else
22:
23: } # while
24:
25: close(WORDS);
26:
27: foreach my $key (sort keys %wordsCount) {
28:     if ( $wordsCount{$key} >= 9 ) {
29:         printf "%-30s => %4d\n", $key => $wordsCount{$key};
30:     } # if
31: } # foreach
```

- Zeile 7-9: Datei öffnen mit Fehlerabfrage
- Zeile 13: Zeile einlesen und \$word zuweisen, \$word auch gleich deklarieren, und überprüfen, ob der Wert auch definiert ist (solche zusammengezogenen Schreibweisen kommen in Perl häufig vor und sind meistens von rechts nach links zu lesen, wenn man da die Klammern weglässt.
- Zeile 16: Wenn ein Wort schon als Hashkey eingetragen ist, dann erhöhe den Value dieses Keys um eins (Zeile 17), sonst erzeuge diesen Key und initialisiere dessen Wert mit **1** (Zeile 20).
- Zeile 27: iteriere über die Keys des Hashes, überprüfe, ob dieses Wort (=Key) öfter als 9 Mal vorgekommen ist (Zeile 28), und wenn ja, dann gib das Wort und die Anzahl formatiert aus (Zeile 29).

Code: hash\_05.pl

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $textfile = 'wordlist.txt';
06:
07: unless (open (WORDS, $textfile)) {
08:     die "Error: can't open '$textfile': $!\n";
09: } # unless
```

```

10:
11: my %wordsCount = ();
12:
13: while (defined (my $word = <WORDS>)) {
14:     chomp($word);
15:     $wordsCount{$word}++;
16: } # while
17:
18: close (WORDS);
19:
20: foreach my $key (sort keys %wordsCount) {
21:     printf("%-30s => %4d\n", $key => $wordsCount{$key})
22:     if $wordsCount{$key} >= 9;
23: } # foreach

```

- Das zweite Beispiel ist eine Kurzschreibweise des ersten; die Zeilen 16-21 wurden durch die Zeile 15 ersetzt, da der ++-Operator, der ja den aktuellen Wert um eins erhöht, auch aus **undef 1** macht.
- In den Zeilen 21/22 wurde das **if** an das Ende gestellt: **printf (...) if Bedingung**; So kann man gut gewichten, was wichtiger ist: das printf oder die Überprüfung der Anzahl. Bei dem nachgestellten **if** kann man auf die Klammern um die Bedingung herum verzichten. Bei dem nachgestellten **if** kann man jedoch kein **elsif** oder **else** verwenden.

### Sortierungen:

Bisher haben wir die Funktion **sort()** immer benützt, wenn wir was asciibetisch sortiert haben wollten. Manchmal will man jedoch, daß unabhängig von der Groß- und Kleinschreibung sortiert wird, oder numerisch (10 ist numerisch gesehen größer als 2). Dafür kann man der Funktion sort einen Block mitgeben, der einen Vergleichsalgorithmus implementiert, der dann von **sort** verwendet wird. Die beiden Elemente, die da verglichen werden sollen, stehen in den globalen Variablen \$a und \$b (deshalb sollte man die auch sonst im Programm nicht verwenden, weil sie nämlich von der Überprüfung von **use strict**; ausgenommen sind).

Asciibetisch bedeutet, nach den ASCII-Werten von Zeichen sortieren, z.B. A => 65, Z => 90, a => 97, z => 122

Bisher haben wir sort immer ohne diesen Block benützt, was der folgenden Schreibweise entsprechen würde:

```
@listeSortiert = sort { $a cmp $b } @liste;
```

cmp vergleicht zwei Zeichenketten (**\$a** und **\$b**) nach ASCII-Zeichen, und liefert **-1** zurück, wenn **\$a** kleiner ist; **0** wenn beide gleich groß sind und **+1**, wenn **\$b** kleiner ist. Sort wertet diese Ergebnisse (negativ, 0, positiv) aus und gruppiert diese Elemente in **@listeSortiert** dementsprechend neu.

Wenn wir jetzt so sortieren wollen, daß Gross-/Kleinschreibung keine Rolle spielt, könnten wir folgenderweise schreiben:

```
@listeSortiert = sort { lc($a) cmp lc($b) } @liste;
```

Dadurch würden für den Vergleich beide Kriterien in Kleinschreibung umgewandelt. Natürlich kann man auch in Großschreibung umwandeln, was das identische Ergebnis liefert:

```
@listeSortiert = sort { uc($a) cmp uc($b) } @liste;
```

Wenn man nun Zahlen numerisch sortieren will (1,2,10,20,40), verwendet man anstelle des Zeichenkettenvergleichs **cmp** einfach einen numerischen Vergleich **<=>** :

```
@listeSortiert = sort { $a <=> $b } @liste;
```

Diese Sortierungen waren bisher immer aufsteigend. Wenn man jedoch absteigend sortieren will, kann man entweder noch ein **reverse** einbauen:

```
@listeAbsteigendSortiert = reverse sort { $a <=> $b } @liste;
```

oder besser noch \$a und \$b vertauschen (ist schneller):

```
@listeAbsteigendSortiert = sort { $b <=> $a } @liste;
```

Wenn man z.B. die Keys eines Hash in einer for-Schleife nach den Werten asciibetisch sortiert haben will, könnte man schreiben:

```
foreach my $key (sort { $hash{$a} cmp $hash{$b} } keys %hash) { ... }
```

Wie könnten wir im letzten Beispiel das häufigste Wort als erstes, das zweithäufigste als zweites usw. ausgeben?

### Subroutinen:

In Perl kann man ebenso wie in den meisten anderen Programmiersprachen Codeteile zusammenfassen und in Subroutinen (=Prozeduren oder Funktionen) verlagern, die man dann von außerhalb und innerhalb einer Funktion aufrufen kann. Eine Subroutine ist ein Block, in dem man auch lokale Variablen deklarieren kann. So kann man eine Subroutine bis auf die Parameterübergabe völlig von außen abschotten. Die Verwendung von Subroutinen erspart oft eine Menge Code, weil man die ja an mehreren Stellen eines Programmes aufrufen kann, und verbessert in der Regel die Strukturierung und somit die Lesbarkeit eines Programmes. Und es erleichtert auch die Wiederverwendbarkeit von Code in anderen Programmen.

Eine Subroutine beginnt immer mit dem Schlüsselwort sub, gefolgt von dem Namen der Subroutine und danach folgt der Block, der den Code enthält:

```
sub NameDerSubroutine {  
  # code  
}
```

Folgende Möglichkeiten gibt es, eine Subroutine aufzurufen:

- &NameDerRoutine()
- NameDerRoutine()
- NameDerRoutine
- &NameDerRoutine

Ich empfehle, die erste oder zweite Möglichkeit zu verwenden. Ich selbst verwende immer die erste, weil durch das & vor dem Namen gleich klar wird, daß es sich um eine nicht in Perl eingebaute Subroutine handelt, und weil man auch vor Variablen immer \$ @ oder % schreiben muß. Die erste Möglichkeit umgeht Prototypen. Die dritte Möglichkeit funktioniert nur dann, wenn die Subroutine vorher schon deklariert worden ist. Die vierte Möglichkeit übergibt eine eventuell vorhandene Parameterliste der Funktion an NameDerRoutine weiter.

Code: sub\_1.pl

```
01: #! /usr/bin/perl  
02: use warnings;  
03: use strict;  
04:  
05: &PrintTrennstrich();  
06: print "Hallo, Welt\n";  
07: &PrintTrennstrich();  
08:  
09: sub PrintTrennstrich {  
10:     print "-" x 60, "\n";  
11: } # PrintTrennstrich
```

- Zeile 9-11: Subroutine **PrintTrennstrich** wird erzeugt
- Zeile 10: der x-Operator "multipliziert" die Zeichenkette, die vorher steht, durch den Wert, der dahinter steht; in diesem Fall wird also ein Strich sechzig Mal aufgegeben
- Zeile 5 und 7: Die Subroutine **PrintTrennstrich** wird aufgerufen

Output:

```
F:\apacheweb\test_8085\html\codes>sub_1.pl
-----
Hallo, Welt
-----
F:\apacheweb\test_8085\html\codes>
```

Mit der Funktion `return(...)` kann man Werte aus der Subroutine als Liste ans Hauptprogramm zurückgeben, z.B.

Code: `sub_2.pl`

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: print &GetTrennstrich();
06: print "Hallo, Welt\n";
07: print &GetTrennstrich();
08:
09: sub GetTrennstrich {
10:     my $string = "-" x 60;
11:
12:     return ($string . "\n");
13: } # GetTrennstrich
```

Bei der Rückgabe von Listen und Hashes muß man darauf achten, daß die bei `return` in eine Liste umgewandelt werden und man somit eventuell die Listen nicht mehr eindeutig aus der Übergabeliste extrahieren kann:

```
sub Subroutine {
    my @list1 = (A..F);
    my @list2 = (G..M);
    return (@list1, @list2);
} # Subroutine
my (@list1, @list2) = &Subroutine();
```

Hier stehen in `@list1` alle Werte A-M und `@list2` ist leer. Genauso beim Hash, der da on-the-Fly in ein Array umgewandelt wird.

### Referenzen:

Um dieses Problem lösen zu können, muß man sich entweder selbst um die Verwaltung der Rückgabeparameterliste kümmern, oder man verwendet Referenzen. Eine Referenz ist ein neuer Variablentyp. Es handelt sich dabei um einen Verweis auf eine andere Variable (d.h. eigentlich auf den Speicherbereich einer Variable), genauso wie in Java und ähnlich wie ein Pointer in C/C++. Eine Referenz ist auch eine skalare Variable, d.h. alle Namensregeln einer skalaren Variablen treten dort in Kraft:

- der Name einer Referenz beginnt stets mit einem `$`
- danach muß ein Buchstabe oder Unterstrich folgen
- danach null oder mehr Buchstaben, Unterstriche oder Ziffern
- Umlaute wie ä, ö, Ü sind nicht erlaubt

### Array-Referenzen:

- Eine Referenz auf eine Variable wird folgendermaßen erzeugt: `\@array`
- Der Inhalt einer Arrayreferenz kann folgendermaßen dereferenziert werden: `@{ $arrayRef }`.
- In Perl kann man die geschwungenen Klammern häufig weglassen: `@$arrayRef`
- Auf ein Element einer Arrayreferenz kann man folgendermaßen zugreifen: `$arrayRef->[ $index ]` (oder die alte Schreibweise: `_${arrayRef}[ $index ]` oder `$$arrayRef[ $index ]`; das `$` ersetzt in diesem Fall das `@`, weil

in Perl immer das Zeichen für den Typ gesetzt wird, der herauskommt. Und da auch ein Element eines Arrays ein Skalar ist, muß man da `${ }[$index]` schreiben).

Code: `ref_1.pl`

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my @array = (0..9);
06: print "Liste: ", join(" ", @array), "\n";
07:
08: my $arrayRef = \@array;
09: print "ListenReferenz: ", join(" ", @{$arrayRef} ), "\n\n";
10:
11: $array[1] = 10; # das Array aendern
12: print "Aenderung1: ", join(" ", @array), "\n";
13: print "Aenderung1: ", join(" ", @$arrayRef), "\n\n";
14:
15: $arrayRef->[3] = 29; # einen Wert der Daten hinter der Referenz aendern
16: print "Aenderung2: ", join(" ", @array), "\n";
17: print "Aenderung2: ", join(" ", @$arrayRef), "\n";
18:
```

Output:

```
F:\apacheweb\test_8085\html\codes>ref_1.pl
Liste: 0 1 2 3 4 5 6 7 8 9
ListenReferenz: 0 1 2 3 4 5 6 7 8 9

Aenderung1: 0 10 2 3 4 5 6 7 8 9
Aenderung1: 0 10 2 3 4 5 6 7 8 9

Aenderung2: 0 10 2 29 4 5 6 7 8 9
Aenderung2: 0 10 2 29 4 5 6 7 8 9

F:\apacheweb\test_8085\html\codes>
```

### Hash-Referenzen:

- Eine Referenz auf eine Variable wird folgendermaßen erzeugt: `\%hash`
- Der Inhalt einer Hashreferenz kann folgendermaßen dereferenziert werden: `%{ $hashRef }`.
- In Perl kann man die geschwungenen Klammern häufig weglassen: `%( $hashRef )`
- Auf ein Element einer Hashreferenz kann man folgendermaßen zugreifen: `$hashRef->{ $key }` (oder die alte Schreibweise: `$( $hashRef ){ $key }` oder `$$hashRef{ $index }`; das `$` ersetzt in diesem Fall das `%`, weil in Perl immer das Zeichen für den Typ gesetzt wird, der herauskommt. Und da auch ein Wert eines Hashes ein Skalar ist, muß man da `$( ){ $key }` schreiben).

Code: `ref_2.pl`

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my %hash = ( key1 => 'value1',
06:              key2 => 'value2',
07:              key3 => 'value3',
08:              );
09:
10: print "Hash: ";
11: foreach my $key ( keys %hash ) { print "$key => $hash{ $key } "; }
12: print "\n";
13:
```

```

14: my $hashRef = \%hash;
15: print "HashRef: ";
16: foreach my $key (keys %$hashRef) {    print "$key => $hashRef{$key} "; }
17: print "\n";
18:
19: $hashRef->{key4} = 'value4'; # neues key/value-Paar anlegen
20: print "Aenderung1: ";
21: foreach my $key (keys %hash) {    print "$key => $hash{$key} "; }
22: print "\n";
23: print "Aenderung1: ";
24: foreach my $key (keys %$hashRef) {    print "$key => $hashRef{$key} "; }
25: print "\n";
26:
27: $hash{key1} = 'perl'; # Wert aendern
28: print "Aenderung2: ";
29: foreach my $key (keys %hash) {    print "$key => $hash{$key} "; }
30: print "\n";
31: print "Aenderung2: ";
32: foreach my $key (keys %$hashRef) {    print "$key => $hashRef{$key} "; }
33: print "\n";

```

Output:

```

F:\apacheweb\test_8085\html\codes>ref_2.pl
Hash: key2 => value2  key1 => value1  key3 => value3
HashRef: key2 => value2  key1 => value1  key3 => value3
Aenderung1: key2 => value2  key4 => value4  key1 => value1  key3 => value3
Aenderung1: key2 => value2  key4 => value4  key1 => value1  key3 => value3
Aenderung2: key2 => value2  key4 => value4  key1 => perl  key3 => value3
Aenderung2: key2 => value2  key4 => value4  key1 => perl  key3 => value3

F:\apacheweb\test_8085\html\codes>

```

Es gibt noch weitere Referenzen:

- Referenzen auf Zahlen oder Zeichenketten (da ist das Dereferenzierungszeichen \$): treten fast nie auf
- Referenzen auf Subroutinen (Dereferenzierungszeichen &): werden im Fortgeschrittenenkurs behandelt
- Referenzen auf Reguläre Ausdrücke: kommen sehr selten in der objektorientierten Programmierung vor
- GLOBs: (Erkennungszeichen: \*): sind ein Überbleibsel aus dem alten Perl4, als es noch keine Referenzen gab und kommen heutzutage höchstens noch bei sehr fortgeschrittenen Handles (z.B. tief verborgen in der Netzwerksocketprogrammierung) vor

Um zu erkennen, von welchem Typ eine Referenz ist, kann man die Funktion `ref($xRef)` verwenden.

```
print ref( $referenz );
```

Die Ergebnisse können sein: ARRAY, HASH, SCALAR, CODE, REF, ...

### Komplexere Rückgabewerte aus Subroutinen:

Code: `sub_3.pl`

```

01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my ($value1, $arrayRef, $hashRef) = &GetData();
06: my @array = @$arrayRef;
07: my %hash = %$hashRef;
08:
09: # -----

```

```

10: sub GetData {
11:     my $string = "abcde";
12:     my @array = (1..10);
13:     my %hash = ( a => 1, b => 2, c => 3 );
14:
15:     return ($string, \@array, \%hash);
16: } # GetData

```

- Zeile 15: der String wird 1:1 der Rückgabeliste hinzugefügt, das Array und der Hash jeweils als Referenz
- Zeile 5: der String, die Arrayreferenz und die Hashreferenz werden aus der Rückgabeliste ausgelesen
- Zeile 6: Dereferenzierung der Arrayreferenz
- Zeile 7: Dereferenzierung der Hashreferenz

### Parameterübergabe an Subroutine:

Code: sub\_4.pl

```

01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: # Ueberschrift ausgeben
06: printf ("%4s | %-8s\n", 'Zahl', 'Wurzel');
07: print '-' x 15, "\n";
08:
09: # Werte ausgeben
10: foreach my $i (1..10) {
11:     my $wurzel = &BerechneWurzel($i);
12:     printf "%4i | %2.6f\n", $i, $wurzel;
13: } # foreach
14:
15: # -----
16: sub BerechneWurzel {
17:     my ($zahl) = @_;
18:
19:     my $wurzel = sqrt($zahl);
20:
21:     return $wurzel;
22: } # GetData

```

- Zeile 11: Die Subroutine BerechneWurzel wird aufgerufen und ihr der Parameter `$i` übergeben
- Zeile 17: Der Parameter `$zahl` (`= $i`) wird aus der Parameterliste `@_` ausgelesen.
- Zeile 19: Die Quadratwurzel von `$zahl` wird berechnet
- Zeile 20: und wieder zurückgegeben und in Zeile 11 an die Variable `$wurzel` übergeben
- Zeile 12: `$i` und `$wurzel` werden formatiert ausgegeben

Hier nochmal das Beispiel mit den Hashreferenzen, diesmal aber mit Subroutinen.

Code: sub\_5.pl

```

01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my %hash = ( key1 => 'value1',
06:             key2 => 'value2',
07:             key3 => 'value3',
08:             );
09: &PrintHash('Anfang', %hash);
10:

```

```

11: my $hashRef = \%hash;
12: &PrintHashRef('Anfang', $hashRef);
13:
14: $hashRef->{key4} = 'value4'; # neues key/value-Paar anlegen
15: &PrintHash('Aenderung1', %hash);
16: &PrintHashRef('Aenderung1', $hashRef);
17:
18: $hash{key1} = 'perl'; # Wert aendern
19: &PrintHash('Aenderung2', %hash);
20: &PrintHashRef('Aenderung2', $hashRef);
21:
22: # -----
23: sub PrintHash {
24:     my ($prefix, %data) = @_;
25:
26:     print "HASH: $prefix: ";
27:     foreach my $key (keys %data) {
28:         print "$key = $data{$key} ";
29:     } # foreach
30:     print "\n";
31: } # PrintHash
32: # -----
33: sub PrintHashRef {
34:     my ($prefix, $data) = @_;
35:
36:     print "HRef: $prefix: ";
37:     foreach my $key (keys %$data) {
38:         print "$key = $data->{$key} ";
39:     } # foreach
40:     print "\n";
41: } # PrintHashRef

```

Wenn man nur das erste Element aus der Übergabeparameterliste braucht:

```
my $param1 = shift(@_); # lange Schreibweise
```

```
my $param1 = shift(); # Abkürzung
```

```
my $param1 = shift; # ganz kurz
```

sonst müßte man schreiben: `my ($param1) = @_;` weil ein Array im skalaren Kontext sonst die Anzahl seiner Elemente zurückgibt (vergleiche `$anzahl = scalar(@liste);`);

### Reguläre Ausdrücke:

Die Suche von Zeichenketten mit den Funktionen (r)index oder eq ist nicht besonders mächtig, dafür aber sehr umständlich. Und Perl wurde ja dazu entwickelt, damit man einfache Probleme einfach lösen kann, und auch komplexe Probleme lösbar zu machen. Und deshalb gibt es da viel mächtigere Mittel, die man reguläre Ausdrücke nennt. Diese Bezeichnung kommt aus der Mathematik und haben in viele Programmiersprachen, Editoren und Kommandozeilenwerkzeuge Einzug gehalten, z.B. Elisp, Python, Tcl, emacs, sed, grep, find, usw. Leider gibt es da auch verschiedene Dialekte, die zwar alle auf einer Basis aufbauen, aber doch manchmal unterschiedliche Prägnanzen haben. Die in Perl verwendeten regulären Ausdrücke sind wahrscheinlich die mächtigsten.

Reguläre Ausdrücke sind äußerst mächtige Hilfsmittel, die jedoch leider auch ziemlich kompliziert sind.

Bei einem regulären Ausdruck handelt es sich um eine Art Grammatik, die auf eine Zeichenkette passt (man sagt auch: matcht) oder nicht. Sie sind so was ähnliches wie die Platzhalter \* und ? in der Dos-Kommandozeile, z.B. werden mit `dir *.txt` im aktuellen Verzeichnis alle sichtbaren Dateien mit der Endung .txt angezeigt. Oder mit `dir *.do?` werden alle sichtbaren Dateien mit den Endungen doa, .dob, .doc, ..., .dot, ... .do5 angezeigt. Mit \* oder ? wird eine kleine Grammatik erzeugt, die da bestimmte Dateinamen erfasst und andere nicht.

Zurück zum Beispiel der Passwortabfrage (Hash):

Code: *hash\_01.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my %passwords = (
06:     "dumbledore" => 'geheim',
07:     'ron'         => "weasly",
08:     "harry"      => "potter",
09:     "neville"   => '',
10: );
11:
12: print "Usernamen eingeben: ";
13: my $username = <STDIN>; chomp($username);
14:
15: print "Passwort eingeben: ";
16: my $password = <STDIN>; chomp($password);
17:
18: if (exists $passwords{$username} ) {
19:
20:     if ( $password eq $passwords{$username} ) {
21:         print "Passwort korrekt. Herzlich willkommen, $username\n";
22:     } # if
23:
24:     else {
25:         print "Passwort $passwords{$username} ist nicht korrekt\n";
26:     } # else
27:
28: } # if
29:
30: else {
31:     print "Kenne niemanden mit dem Usernamen $username\n";
32: } # else
33:
```

Bei diesem Beispiel mußte man `harry` eingeben; `Harry` wurde z.B. nicht erkannt. Damit da Groß- und Kleinschreibung ignoriert wird, könnte man mit einem regulären Ausdruck folgendermaßen schreiben:

Code: *regex\_01.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my %passwords = (
06:     "dumbledore" => 'geheim',
07:     'ron'         => "weasly",
08:     "harry"      => "potter",
09:     "neville"   => '',
10: );
11:
12: print "Usernamen eingeben: ";
13: chomp(my $username = <STDIN>);
14:
15: foreach my $key (keys %passwords) {
16:
17:     if ($username =~ m/$key/i) {
18:         print "Der Username '$key' wurde gefunden\n";
19:
20:     } # if
21: } # foreach
22:
```

- Zeile 13 ist wieder eine Zusammenziehung von mehreren Kommandos

Zeile 17: ein regulärer Ausdruck hat immer die folgende Form:

\$zeichenkette =~ m/Muster/Optionen;

\$zeichenkette ist der String, in dem gesucht wird

=~ ist der Operator; auf Deutsch sagt man dafür: passt auf, matcht auf; in Englisch: matches, is like

m/.../ umrahmt den Ausdruck. Anstelle von / können auch andere Trennzeichen verwendet werden, z.B.

m(...), m|...|, m#...#, m~...~, m{...}

Wenn das Trennzeichen jedoch / ist, kann man das m auch weglassen: /.../

Am Ende kann man noch Optionen angeben, die das Suchverhalten beeinflussen. In diesem Beispiel habe ich m/.../i verwendet, was die Suche von Gross- und Kleinschreibung unabhängig durchführt.

Unser Vergleich in dem letzten Beispiel hat jedoch eine Schwäche: was passiert, wenn da als Username z.B. **ronald** oder **charon** angegeben wird?

### Ankerwerfen:

Dieses Problem kann man durch die Verwendung von Ankern lösen:

^ ist ein Anker, der am Anfang der Zeichenkette passt

\$ ist ein Anker, der am Ende der Zeichenkette passt.

Mit diesen Hilfsmitteln können wir die Regex (=Kurzausdruck für regulären Ausdruck) sicherer machen:

```
if ($username =~ m/^[key$/i ) {
```

### Sonderzeichen in der Regex:

In einer Regex werden genauso wie in einer Zeichenkette mit doppelten Anführungszeichen bestimmte Zeichen(-kombinationen) (z.B. \, \n, \t, \$) zu Sonderzeichen. Wenn man in einer Zeichenkette ein \$ ausgeben will, muß man entweder einfache Anführungszeichen verwenden, oder vor dieses Zeichen einen Backslash stellen ( \\$ steht für das Zeichen \$). Genauso bei den Formatanweisungen von (s)printf: da kann man ein %-Zeichen angeben, indem man % schreibt.

Wenn man will, daß \$key nicht auf den Inhalt der Variablen \$key passt, sondern auf die Zeichen von \$key, dann muß man das \$ escapen, d.h. einen Backslash davor schreiben: m/^\\$key\$. Oder wenn man will, daß ein \$ am Ende des Strings als \$ erkannt wird, muß man genauso schreiben: m^\\$/; Weiters haben noch die Zeichen ? + \* [ - / \ ( Sonderbedeutungen

## Grammatik:

Verwenden wir mal folgendes Beispiel als Basis, bei dem sich nur das Muster ändert:

Code: `regex_02.pl`

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my @sprachen = qw(Perl Pascal VisualBasic Quickbasic
06:                 Fortran C C++ Java Python Smalltalk
07:                 Cobol Oberon Basic Delphi Ada);
08:
09: foreach my $sprache (@sprachen) {
10:     if( $sprache =~ m/r/ ) {
11:         print "$sprache\n";
12:     } # if
13: } # foreach
```

In diesem Beispiel ändern wir derzeit nur das Muster in Zeile 10:

1. Im aktuellen Beispiel geben wir alle Programmiersprachen aus, die ein kleines **r** beinhalten (Perl und Fortran)
2. Wir wollen alle Sprachen wissen, die mit **P** beginnen und mindestens 5 Zeichen lang sind:  
Das Sonderzeichen **.** (der Punkt) steht für genau ein beliebiges Zeichen (außer **\n**). Also könnten wir schreiben:  
`$sprache =~ m/^\P...../;` (Pascal, Python)  
(Falls man wirklich einen Punkt haben will, muß man den escapen und schreiben: `\.`)
3. alle Sprachen, in denen zu Beginn ein **P** steht und am Ende ein **l** und mindestens ein Zeichen dazwischen ist:  
Mit dem Sonderzeichen **+** kann man sagen, daß der letzte Ausdruck einmal oder öfter kommen muß:  
`$sprache =~ m/^\P.+l$/;` (Perl, Pascal)
4. alle Sprachen, die am Ende ein **++** haben: Da es sich bei dem **+** um ein Sonderzeichen handelt, muß man schreiben  
`$sprache =~ m/^\++$/i;`  
Mit `{Anzahl}` kann man auch genau angeben, wie oft das letzte Zeichen kommen soll:  
`$sprache =~ m/^\++{2}$/i;`
5. alle Sprachen, deren Länge mindestens 6 und höchstens 11 Zeichen beträgt: Mit `{start,ende}` kann man auch Bereiche angeben  
`$sprache =~ m/^\.{6,11}$/i;`
6. alle Sprachen, deren Länge höchstens 5 Buchstaben beträgt:  
`$sprache =~ m/^\.{0,5}$/i;`  
+ könnte man auch als `{1,}` schreiben (von 1 bis unendlich)  
`.*` steht für `{0,}`, also nullmal oder öfter
7. alle Sprachen, die auf **i** oder **ic** enden:  
`$sprache =~ m/ic{0,1}$/i;`  
Für den Ausdruck `{0,1}` gibt es auch eine Abkürzung: `?`  
`$sprache =~ m/ic?$/i;`
8. alle Sprachen, in denen ein großes oder kleines **B** vorkommt, gefolgt von einem kleinen **a**: Mit `[Bb]` kann man Zeichenklassen angeben, von denen genau eins vorkommen muß:  
`$sprache =~ m/[Bb]a/;`
9. alle Sprachen, die nur aus den Buchstaben von **A** bis **O** bestehen:  
`$sprache =~ m/^[abcdefghijklmnop]+$/i;`  
Mit `-` kann man auch Bereiche angeben, z.B.  
`$sprache =~ m/[a-o]+$/i;`  
Und wenn man auf das `/i` auch noch verzichten will, kann man auch schreiben:  
`$sprache =~ m/^[a-oA-O]+$/;`

<b>Ausdruck :</b>	<b>passt auf:</b>
<code>^</code>	den Anfang einer Zeichenkette
<code>\$</code>	das Ende einer Zeichenkette
<code>.</code>	ein beliebiges Zeichen (außer <code>\n</code> )
<code>\.</code>	einen Punkt
<code>a{5}</code>	fünf a hintereinander
<code>a{4,}</code>	mindestens vier a hintereinander
<code>a{0,6}</code>	höchstens sechs a hintereinander
<code>a{3,5}</code>	mindestens drei a, aber maximal 5 a
<code>a+</code>	mindestens ein a
<code>a*</code>	kein, ein oder mehrere a
<code>[a-z]</code>	einen Kleinbuchstaben (=Zeichenklasse)
<code>[^a-z]</code>	ein Zeichen, das kein Kleinbuchstabe ist
<code>[0-9]</code>	eine Ziffer
<code>\d</code>	eine Ziffer
<code>[^0-9]</code>	ein Zeichen, das keine Ziffer ist
<code>\D</code>	ein Zeichen, das keine Ziffer ist
<code>\s</code>	whitespace (Leerzeichen, Tabulator, Zeilenumbruch)
<code>\S</code>	ein Zeichen, das kein Whitespace ist
<code>\w</code>	ein Zeichen der Zeichenklasse <code>[A-Za-z0-9_]</code>
<code>\W</code>	ein Zeichen, das nicht in der Zeichenklasse <code>[A-Za-z_]</code> ist
<code>\n</code>	ein Zeilenumbruch
<code>\t</code>	ein Tabulator
<code>(abc def)</code>	abc oder def
<code>\b</code>	Wortgrenze (am Anfang oder Ende eines Wortes)
<code>\B</code>	keine Wortgrenze (also in einem Wort)
<code>[\b]</code>	Backspace (lösche das letzte Zeichen; nur in Zeichenklassen, sonst Wortgrenze)
<code>\@</code>	@-Zeichen
<code>\/</code>	/ (wenn das Trennzeichen des Musters / ist, muß es escaped werden)
<code>\?</code>	?-Zeichen

## Speicherung mit regulären Ausdrücken:

Mit ( ) kann man den Text, der durch einen oder mehrere Ausdrücke gefunden wurde, abspeichern. Im Muster selbst kann man mit \1 \2 ... \9 darauf zugreifen, nach dem Muster durch die Spezialvariablen \$1 \$2 ... \$9

Code: *regex\_03.pl*

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $string = "192.168.1.14";
06:
07: if ( $string =~ m/(\d{1,3})\.\(\d+)\.\(\d+)\.\(\d+)/ ) {
08:
09:     print "\$1: $1\n";
10:     print "\$2: $2\n";
11:     print "\$3: $3\n";
12:     print "\$4: $4\n";
13:
14: } # if
15:
```

Es gibt (parallel zu /i) die Option /x, die es erlaubt, Muster so zu schreiben, daß man sie auch wieder lesen kann. Dort kann man beliebig viele Leerzeichen einfügen und sogar Kommentare. Wenn man allerdings in einem Muster nach einem Leerzeichen suchen will, muß man es entweder escapen (also \ ) oder allgemeiner \s verwenden:

Code: *regex\_04.pl*

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $string = "192.168.1.14";
06:
07: if ( $string =~ m/
08:     ( \d+ )      # speichere mindestens eine Zahl
09:     \.          # ein Punkt (nicht speichern)
10:     ( \d+ )      # speichere die naechsten Zahlen
11:     \.          # ein Punkt (nicht speichern)
12:     ( \d{1,3} ) # speichere ein bis drei Zahlen
13:     \.          # ein Punkt (nicht speichern)
14:     ( \d{1,3} ) # speichere die naechsten 1-3 Zahlen
15:     /x ) {
16:
17:     print "\$1: $1\n";
18:     print "\$2: $2\n";
19:     print "\$3: $3\n";
20:     print "\$4: $4\n";
21:
22: } # if
23:
```

Bei komplexeren Suchmustern empfehle ich, diese Möglichkeit zu verwenden, denn dann hat man auch nach Jahren noch die Chance, ein Suchmuster zu verstehen.

## Ersetzungen mit regulären Ausdrücken:

Eine Ersetzung mit regulären Ausdrücken sieht prinzipiell folgendermaßen aus:

```
$string =~ s/Muster/Ersetzung/;
```

z.B. `$string =~ s/\b[Bb]asic/Perl/;`

Code: `regex_05.pl`

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $string = "Seife ist Basich. Heute lernen wir Basic.";
06:
07: $string =~ s/
08:     \b      # wortgrenze
09:     [bB]asic # basic oder Basic
10:     \b      # wortgrenze, damit Basich nicht gefunden wird
11:     /Perl/gx; # ersetze das gefundene durch Perl
12:
13: print "$string\n";
```

- Zeile 11: hier wird die Option /g verwendet, die sagt, daß nach dem ersten Vorkommen noch weiter gesucht werden soll, also alle gefundenen Muster ersetzt werden sollen.

Output:

```
F:\apacheweb\test_8085\html\codes>regex_05.pl
Seife ist Basisch. Heute lernen wir Perl.

F:\apacheweb\test_8085\html\codes>
```

Wenn man will, daß die Ersetzung als Perl-Code ausgeführt wird, kann man als Option /e mitgeben:

```
$string =~ s/(\d\d\d)/ my $i = $1; $i++; "$i$i" /ge;
```

## Reguläre Ausdrücke zu gierig?

Grundsätzlich versucht die Regex-Engine bei Ausdrücken wie `.*` immer, auf möglichst viele Zeichen zu matchen.

Code: `regex_06.pl`

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $string = "Dies ist ein <b>fetter Ausdruck</b>. Und dies ist ein " .
06:     "<b>zweiter fetter Ausdruck</b>. Und dies ein <b>dritter</b>";
07:
08: while ($string =~ m/<b>(.*?)</b>/gs) {
09:     print "$1\n";
10: } # while
```

- Zeile 08: die Option /s sagt Perl, daß `.` auch auf Zeilenumbrüche passen soll

Output:

```
F:\apacheweb\test_8085\html\codes>regex_06.pl
fetter Ausdruck</b>. Und dies ist ein <b>zweiter fetter Ausdruck</b>. Und dies e
in <b>dritter

F:\apacheweb\test_8085\html\codes>
```

Manchmal will man jedoch so wenig Zeichen wie möglich haben. Da kann man nach Quantoren (z.B. +, \*) das Zeichen ? schreiben:

Code: *regex\_07.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $string = "Dies ist ein <b>fetter Ausdruck</b>. Und dies ist ein " .
06:   "<b>zweiter fetter Ausdruck</b>. Und dies ein <b>dritter</b>";
07:
08: while ($string =~ m/<b>(.*?)</b>/gs) {
09:     print "$1\n";
10: } # while
```

Output:

```
F:\apacheweb\test_8085\html\codes>regex_07.pl
fetter Ausdruck
zweiter fetter Ausdruck
dritter

F:\apacheweb\test_8085\html\codes>
```

## Directoryhandling:

Das Öffnen eines Verzeichnisses funktioniert fast genauso wie das Öffnen einer Datei:

Code: *dir\_1.pl*

```
01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $dir = $ENV{'WINDIR'};
06:
07: unless (opendir(DIR, $dir)) {
08:     die "Error in opening '$dir': $!\n";
09: } # unless
10:
11: foreach my $entry (readdir(DIR)) {
12:     if ($entry =~ m/\.exe$/) {
13:         print "$entry\n";
14:     } # if
15: } # foreach
16: closedir(DIR);
```

- Zeile 5: Über die Umgebungsvariable `%ENV` wird das Windows-Verzeichnis ermittelt und in `$dir` gespeichert
- Zeile 7-9: Verzeichnis öffnen und dem Handle `DIR` zuweisen, Fehlerabfrage
- Mit `readdir(DIR)` werden die Inhalte (d.h. Dateien, Verzeichnisse, ...) ausgelesen und bei jeder Iteration `$entry` mit einem neuen Wert versorgt. `readdir()` funktioniert auch kontextabhängig; im skalaren Kontext gibt es einen Eintrag zurück, im Listenkontext alle Einträge.
- Zeile 16: Das Verzeichnis wird wieder geschlossen und der Handle freigegeben
- Zeile 12: Wenn eine Datei die Endung `.exe` hat, gib den Namen aus

## Verzeichnis oder Datei?

Code: `dir_2.pl`

```

01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $dir = $ENV{'WINDIR'};
06:
07: unless (opendir(DIR, $dir)) {
08:     die "Error in opening '$dir': $_\n";
09: } # unless
10:
11: foreach my $entry (readdir(DIR)) {
12:
13:     if (-d "$dir/$entry") {
14:         printf "%-10s $entry\n", 'Verzeichnis';
15:     } # if
16:     elsif (-f "$dir/$entry") {
17:         printf "%-10s $entry\n", 'Datei';
18:     } # elsif
19:     else {
20:         printf "%-10s $entry\n", 'Unbekannt';
21:     } # else
22: } # foreach
23: closedir(DIR);

```

Achtung: bei `opendir(DIR)` muß man den Pfad mit angeben!

### Kommando: Gibt wahr zurück, wenn der Eintrag

- e existiert
- f eine Datei ist
- d ein Verzeichnis ist
- z 0 Bytes groß ist
- s größer als 0 Bytes ist (gibt die Größe zurück)
- T eine Textdatei ist
- B eine Binärdatei ist

## Inhalte filtern:

Entweder verwendet man opendir/readdir/closedir und testet die Dateinamen gegen eine Regex, oder man verwendet glob(Muster):

Code: *dir\_3.pl*

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $dir = $ENV{'WINDIR'};
06:
07: my @files = glob("$dir/*.exe");
08:
09: if ($!) {
10:     die "Fehler: konnte '$dir/*.exe' nicht auslesen: $!\n";
11: } # if
12:
13: foreach my $file (@files) {
14:     print "$file\n";
15: } # foreach
```

- Zeile 7: die Funktion glob kann man mit Dos-Platzhaltern \* ? ansprechen und gibt eine Liste der Dateien zurück
- Zeilen 9-11: Fehler abfangen

## Programmparameter:

Wenn einem Perl-Script Parameter übergeben werden, findet man die in der Liste @ARGV

Code: *params\_1.pl*

```
1: #!/usr/bin/perl
2: use warnings;
3: use strict;
4:
5: print "Programmname: $0\n";
6: print "Parameter:\n";
7: foreach my $argv (@ARGV) {
8:     print "\t$argv\n";
9: }
```

Output:

```
F:\apacheweb\test_8085\html\codes>params_1.pl -c=2 "ausdruck mit leerzeichen"
Programmname: F:\apacheweb\test_8085\html\codes\params_1.pl
Parameter:
    -c=2
    ausdruck mit leerzeichen
F:\apacheweb\test_8085\html\codes>
```

Hier ein Beispiel, wie man Parameter der Form x=y a=b in einen Hash umwandeln kann:

Code: *params\_2.pl*

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my %argv = &ExtractParameters (@ARGV);
06:
07: foreach my $key (sort keys %argv) {
08:     print "$key => $argv{$key}\n";
09: } # foreach
10:
11:
12: # -----
13: sub ExtractParameters {
14:     my @argv = @_;
15:     my %argv = ();
16:
17:     foreach my $argv (@argv) {
18:         my ($key, $value) = split(/=/, $argv, 2);
19:
20:         if (exists $argv{$key}) {
21:             die "Fehler: der Parameter '$key' wurde doppelt angegeben\n";
22:         } # if
23:         else {
24:             $argv{$key} = $value;
25:         } # else
26:
27:     } # foreach
28:
29:     return (%argv);
30: } # ExtractParameters
31: # -----
32:
```

Output:

```
F:\apacheweb\test_8085\html\codes>params_2.pl -c=2 -b=3 -d=4
-b => 3
-c => 2
-d => 4
F:\apacheweb\test_8085\html\codes>
```

### Datum und Zeit:

Die Funktion **time()** liefert die Epochsekunden zurück. Epochsekunden ist die Anzahl der verstrichenen Sekunden seit dem 1.1.1970. Mit **localtime(time)** kann man die Epochsekunden in ein lesbareres Format umwandeln, wobei sich localtime je nach Kontext unterschiedlich verhält.

- Skalarer Kontext: gibt eine Zeichenkette zurück der Form: Wed Sep 3 11:06:10 2002
- Listenkontext: gibt eine Liste zurück, deren Elemente folgende Bedeutung haben:

Position:	Wert:	Bereich:	Anmerkung:
0	Sekunden	0-60	
1	Minuten	0-59	
2	Stunden	0-23	
3	Tag des Monats	1-31	
4	Monat	0-11	Achtung: Januar ist 0
5	Jahre, die seit 1900 vergangen sind	0-138 (oder mehr)	Achtung: dazu immer 1900 addieren
6	Wochentag	0-6	0 ist Sonntag, 1 Montag, ...
7	Tag des Jahres	1-366	
8	Ist Sommerzeit aktiv?	0-1	Wahr, wenn die Sommerzeit aktiv ist

Code: date\_1.pl

```

01: #! /usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $epochSeconds = time;
06: my $printableDate = &ConvertEpochSecondsToDate($epochSeconds);
07:
08: print "EPOCH: $epochSeconds\n";
09: print "String: ", scalar(localtime(time)), "\n";
10: print "Datum: $printableDate\n";
11:
12: # -----
13: sub ConvertEpochSecondsToDate {
14:     my $epochSeconds = shift; # Kurzschreibweise
15:
16:     my @date = ( localtime($epochSeconds) )[0..5];
17:     $date[4]++; $date[5] += 1900;
18:
19:     foreach my $d (@date[0..4]) {
20:         $d = sprintf("%02d", $d);
21:     } # foreach
22:
23:     my $date = join(".", @date[3,4,5]);
24:     my $time = join(":", @date[2,1,0]);
25:
26:     return "$date $time";
27: } # ConvertEpochSecondsToDate

```

Output:

```

F:\apacheweb\test_8085\html\codes>perl date_1.pl
EPOCH: 1062580938
String: Wed Sep 3 11:22:18 2003
Datum: 03.09.2003 11:22:18

F:\apacheweb\test_8085\html\codes>

```

## Datum in Epochsekunden umwandeln:

Code: *date\_2.pl*

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $date = "12.10.2003 16:33:19";
06:
07: if ($date =~ /^
08:     (\d\d?)\.\(\d\d?\)\.\(\d\d\d\d\) # dd.mm.yyyy
09:     \s+
10:     (\d\d?)\:(\d\d?)\:(\d\d?)      # hh:mm:ss
11:     $/x) {
12:
13:     # sek, min, stunde, tag, monat jahr
14:     my @date = ($6, $5, $4, $1, $2, $3);
15:     $date[4]--; # januar ist 0
16:     $date[5] -= 1900; # Jahr 0 ist 1900
17:
18:     use Time::Local; # Modul laden
19:     my $epochSeconds = timelocal(@date); # umwandeln
20:
21:     print "EPOCH: $epochSeconds\n";
22: } # if
23:
24: else {
25:     warn "Kein Datum in $date gefunden\n";
26: } # else
```

## Formate:

Eine der Erklärungen des Namens Perl ist "Practical Extraction and Report Language". Und Perl gibt einem auch gute Hilfe, um Reporte und ähnliches zu erstellen. Bisher haben wir die Funktionen print und printf kennengelernt.

Code: *format\_1.pl*

```
01: #!/usr/bin/perl
02: use warnings;
03: use strict;
04:
05: my $inputFile = "rangliste.txt";
06: my @lines = ();
07:
08: unless (open (INPUT, $inputFile)) {
09:     die "Error: couldn't read '$inputFile': $!\n";
10: } # unless
11:
12: # lese die Zeilen ein
13: foreach my $line (<INPUT>) {
14:     chomp($line);
15:     next if $line =~ m/^\s+$/; # ueberspringe Leerzeilen
16:     push (@lines, $line); # fuege sie zu @lines hinzu
17: } # foreach
18:
19: close(INPUT);
20:
```





## Installation von Perl-Modulen:

Perl-Module findet man unter folgenden URLs:

1. Für ActivestatePerl (<http://www.activestate.com/>) gibt es einige schon vorkompilierte Module, die sehr einfach zu installieren sind. Unter Windows sollten diese verwendet werden (wenn vorhanden):
  - o Perl5.8: <http://ppm.activestate.com/PPMPackages/zips/8xx-builds-only/>
  - o Perl5.6: <http://ppm.activestate.com/PPMPackages/zips/6xx-builds-only/>
  - o Perl5.5: <http://ppm.activestate.com/PPMPackages/zips/5xx-builds-only/>

Diese Module können als Zip-Dateien downgeloaded und z.B. mit Winzip entpackt werden. Danach wechsle man in das Verzeichnis, in dem eine Datei mit der Endung *Modulname.ppd* (*Modulname* verwende ich als Platzhalter) liegt und gibt dort ein: **ppm install Modulname.ppd**. Dieser Weg ist ideal, wenn ein Rechner keine Internetverbindung hat, oder wenn diese über einen komplexeren Proxy abgewickelt wird.

Bei aktiver Internetverbindung kann man diese Module auch mit ppm direkt aus dem Internet installieren:

```
E:\>ppm
PPM interactive shell (2.1.6) - type 'help' for available commands.
PPM> install Acme-Code-Police
Install package 'Acme-Code-Police?' (y/N): y
Installing package 'Acme-Code-Police'...
Bytes transferred: 1226
PPM> quit
Quit!

E:\>
```

Mit **search Modulname** bekommt man eine Liste aller Module, deren Namen **Modulname** enthalten. Bei der Online-Installation mit ppm braucht man sich um die Perl-Version keine Gedanken machen, weil ppm automatisch nach den richtigen Modulen sucht.

Activestate-Perl und vorcompilierte ppm-Module gibt es derzeit für Win32, Linux und Solaris.

2. Wenn diese Vorgehensweise nicht zum Ziel führt oder man kein Activestate-Perl benützt, muß man die Module selbst compilieren. Dazu gibt es einfachere und schwierigere, aber sicherere Wege:
  - o Manuelle Installation: Modul von CPAN (=Central Perl Architecture Network, die zentrale Modulanlaufstelle für Perl auf <http://www.cpan.org/>) downloaden (als .tar.gz oder .zip), das Archiv entpacken und sich die Datei namens README (und, wenn vorhanden, INSTALL) durchlesen. Da steht drinnen, wie man das Modul installieren muß. Im Laufe der Zeit hat sich jedoch ein Standardweg herausgebildet, dem mittlerweile fast alle CPAN-Module folgen. Dafür braucht man ein **make** für das entsprechende Betriebssystem, und für manche Module auch einen C-Compiler. (Für Windows bekommt man ein Standalone-make z.B. von <ftp://ftp.microsoft.com/Softlib/MSLFILES/nmake15.exe> . Downloaden und in ein Verzeichnis entpacken, das in PATH eingetragen ist, z.B. in das Verzeichnis Perl/bin. Danach lautet der Standardweg:

```
... in das Verzeichnis mit der Makefile.PL wechseln
perl Makefile.PL
make
make test
make install
```

Bei Windows muß anstelle von **make nmake** geschrieben werden. Und nur wenn ein Befehl fehlerfrei durchläuft, kann der nächste ausgeführt werden (es macht wohl ziemlich wenig Sinn, ein fehlerhaftes Modul zu installieren, oder?) Wenn da eine Fehlermeldung wie **cl.exe not found** ausgegeben wird, benötigt man MsVisualC++6.

- o Da dieser Standardweg mittlerweile bei fast allen Modulen eingehalten wird, wurde dieser Weg zu einem Modul zusammengefasst, das einem die Installation erleichtert. Dieses Modul nennt sich CPAN, und wird in der Shell folgendermaßen ausgeführt:

```
E:\>perl -MCPAN -e shell
cpan shell -- CPAN exploration and modules installation (v1.61)

cpan> install HTML::Template
# ..... gekuerzt ...
```

```
cpan> quit
E:\>
```

Wenn dieser Weg nicht funktioniert, kann man noch den manuellen Weg (wie oben beschrieben) wählen.

Damit der Weg mit CPAN auch unter Windows gut funktioniert, braucht man neben (**n**)**make** noch einige weitere Programme, die unter Linux/Unix schon standardmäßig installiert sind (bzw. sein sollten). Von <http://www.cygwin.com/> kann eine komplette Kommandozeilenumgebung (bash) für Windows downgeloaded und installiert werden, die fast alle nötigen Programme für Windows enthält. Dies sollte man vor der ersten Verwendung von CPAN installieren, weil dieses Modul beim ersten Aufruf eine ganze Menge Fragen stellt, von denen man fast immer die Default-Antworten verwenden kann (nur darauf achten, dass **nmake** anstelle des **cygwin-make's** verwendet wird). Die **cygwin-Tools** sind auch für den nächsten Weg eine Voraussetzung:

- o **CPANPLUS**: Dieses Modul ist leider noch kein Standardmodul, wird es aber in einer der nächsten Perl-Versionen werden. Es versucht recht erfolgreich, die Probleme mit CPAN zu lösen und die Modulsuche und -installation komfortabler zu gestalten.

```
E:\>perl -MCPANPLUS -e shell
CPANPLUS::Shell::Default -- CPAN exploration and modules installation
*** Please report bugs to <cpanplus-bugs@lists.sourceforge.net>.
*** Using CPANPLUS::Backend v0.042.  ReadLine support disabled.
CPAN Terminal> i TestCPAN Terminal>
Installing: Test
Checking if your kit is complete...
Looks good
Writing Makefile for Test
  E:\apps\gnu\Perl58\bin\perl.exe "-MExtUtils::Command::MM" "-e" \
"test_harness(0, 'blib\lib', 'blib\arch')" t\fail.t t\mix.t t\onfail.t t\
qr.t t\skip.t t\
success.t t\todo.t
t\fail.....ok
t\mix.....ok
t\onfail.....ok
t\qr.....ok
t\skip.....ok
t\success.....ok
      1/11 skipped: just testing skip()
t\todo.....ok
All tests successful, 1 subtest skipped.
Files=7, Tests=40,  0 wallclock secs ( 0.00 cusr +  0.00 csys =  0.00 CPU)

Microsoft (R) Program Maintenance Utility Version 6.00.8168.0
Copyright (C) Microsoft Corp 1988-1998. All rights reserved.

Successfully installed Test

All modules installed successfully
CPAN Terminal> help
# .... gekuerzt ...
CPAN Terminal> q
Exiting CPANPLUS shell
E:\>
```

Wenn der Weg über **ppm** nicht möglich oder erfolgreich ist, empfehle ich den **CPANPLUS**-Weg. Die Shell kann meistens auch über den Befehl **cpalp** aufgerufen werden... Die Hilfe zur Shell kann man sich mit dem Befehl **help** ausgeben lassen. Die sollte man sich unbedingt mal durchlesen, weil darin steht, wie man sich viel Tipperei ersparen kann.

## Dokumentation zu Perl-Modulen:

Fast alle Module haben die Dokumentation eingebaut (im POD-Format). Diese Dokumentation kann man sich, wenn das Modul installiert ist, mit dem Kommando **perldoc Modulname** ansehen, z.B.

```
E:\>perldoc CGI
NAME
    CGI - Simple Common Gateway Interface Class
SYNOPSIS
    # CGI script that creates a fill-out form
    # and echoes back its values.
# ... gekuerzt ...
E:\>
```

Wenn das Modul noch nicht installiert ist, kann man bei <http://search.cpan.org/> danach suchen.

## Gute Perl-Bücher für Anfänger und Fortgeschrittene:

Titel:	Autor(en):	Verlag:	Preis:	Anmerkung:
Einführung in Perl	Randal L. Schwartz, Tom Phoenix	O'Reilly	~35€	Gute Einführung in Perl für Unix/Linux-Betriebssysteme.
Einführung in Perl für Win32-Systeme	Randal L. Schwartz, Erik Olson, Tom Christiansen	O'Reilly	~20€	Gute Einführung in Perl für Windows-Betriebssysteme
Programmieren mit Perl	Larry Wall, Tom Christiansen, Jon Orwant, Randal Schwartz	O'Reilly	~56€	Das Kamel-Buch ist eine vollständige Referenz zu Perl. Wenn man all das kann, was darin vorkommt, kann man Perl sehr gut. Man sollte jedoch mindestens die dritte Auflage kaufen, die zweite ist nur teilweise noch Up-To-Date
Perl-Kochbuch	Tom Christiansen, Nathan Torkington	O'Reilly	~46€	Kleine und große Rezepte zu Perl und eine sehr gute Ergänzung zu Programmieren mit Perl oder Einführung in Perl. Sollte jeder Perl-Programmierer haben.
Programmierung mit Perl DBI	Alligator Descartes, Tim Bunce	O'Reilly	~38€	Datenbankprogrammierung mit Perl (SQL über ODBC, DBI, ...)
Effektiv Perl programmieren	Joseph N. Hall, Randal L. Schwartz	Addison-Wesley	~20€	Guter Leitfaden, wie man besser und effektiver Perl-Programmieren kann

## Hilfe bei Perl-Problemen:

Es gibt mehrere Perl-Webforen, in denen man gute Hilfe findet, wenn man sein Problem genau beschreibt und auch Informationen wie Fehlermeldungen, welches Betriebssystem verwendet wird usw. angibt.

- [www.perlmonks.org](http://www.perlmonks.org) : Das größte Forum weltweit, wo man sogar Antworten von Perl-Größen wie Randal Schwartz (=merlyn) oder Damian Conway (=TheDamian) bekommen kann. Sprache: Englisch. Dort bin ich unter dem Pseudonym strat vertreten, allerdings nicht besonders aktiv.
- [www.perl-community.de](http://www.perl-community.de) : Das größte deutschsprachige Perl-Forum, das nach seinem Umzug von <http://www.perl.de/> wieder neu aufgebaut wurde. Dort ist das Niveau auch recht gut. Dieses Forum ist eins meiner größten Hobbies. Dort bin ich unter dem Pseudonym Strat zu finden.

**Perl-Stammtische:**

In fast jeder größeren Stadt gibt es Perl-Mongers-Gruppen, die sich meistens 1-2 Mal pro Monat treffen, Vorträge über Perl oder Perl-verwandte Themen halten, gemeinsam Projekte machen oder einfach gemütlich ein Bier trinken. Auf <http://www.pm.org/groups/europe.html> sind die europäischen Perlmongers-Gruppen aufgelistet. Ich bin Mitglied bei den Frankfurter Perl-Mongers (<http://frankfurt-pm.org/>) und Röderbergweg.pm.

**Veranstaltungen:**

Jedes Jahr von Aschermittwoch bis zum darauffolgenden Freitag findet der Deutsche Perlworkshop mit Tutorials und Vorträgen statt: <http://www.perlworkshop.de/>

Jedes Jahr findet irgendwo in Europa die YAPC::Europe statt, z.B. 2004 in Belfast: <http://www.yapc.org/Europe/>